

GAMS による応用一般均衡分析： 海外部門のあるモデル

小 平 裕

1. はじめに
2. 貿易の定式化
3. モデルの方程式体系
4. 基準均衡データセット
5. GAMS 入力ファイル
6. 結果のまとめ
7. 残された課題

1. はじめに

本稿の目的は、前稿（小平 (2002d)）で構築した age 4 モデルを拡張して、海外部門を取り入れ貿易が行われるモデルを完成させることである。われわれは、数理計画法解法プログラム GAMS を利用した応用一般均衡分析を習得するために、基本モデル（age 1 モデル，小平 (2002a)）を構築し、段階的に拡張してきた（age 2 モデル（小平 (2002b)）では中間投入，age 3 モデル（小平 (2002c)）では資本市場，age 4 モデル（小平 (2002b)）では政府部門を追加）。本稿は、小平 (2002a) 以来進めてきた一連の拡張の最終段階にあたる。

国内経済（自国）と海外部門 *RoW* との間で行われる取引の形態には、商品の輸入と輸出，所得（購買力）の移転の 3 種類がある。このうち、商品の輸入と輸出において自国がどの商品を輸出しどの商品を輸入するかの決定，すなわち二国間の貿易パターンの決定は、理論的には比較優位の原理に基づいて説明されるが、1 つの国がある商品を同時に輸出し輸入することはない。それにも関わらず、応用モデルではある商品がある国から同

時に輸出され輸入される現象がよく観察されており、双方向貿易（双方向輸送）¹⁾と呼ばれている。

本稿の狙いの1つは、この現象が観察されるようなモデルを構築して、国際価格の変化に対する国内価格や生産、消費の反応をより現実的に説明することができるようにすることである。双方向貿易は、採用される商品集計水準（財分類）の粗さ、物理的品質の違い、入手可能な時間や場所の違いなどに起因すると説明されているので、本稿では国内で生産され国内で販売される国産財、国内で生産され海外に販売される輸出財、そして海外で生産され国内で販売される輸入財の間には、これらの要因に基づく不完全代替性と不完全変形可能性があると想定する。すなわち、どちらも国内の財市場で取引される商品であるが、国内で生産される国産財と海外で生産される輸入財の間には、国内の需要者達から見ると不完全代替性があり、またどちらも自国の生産者達によって国内生産される商品であるが、国内市場に供給される国産財と海外に供給される輸出財の間には不完全変形可能性があると仮定する。このように不完全代替性と不完全変形可能性を仮定することによって、モデル構造を大幅に修正することなく、双方向貿易という現象をモデルの中で処理することを可能にする。

2. 貿易の定式化

前稿（小平（2002d））の age 4 モデルからの大きな変更点は貿易の導入であるが、応用モデルでは輸出と輸入の取扱いに関連して注意を要する点がある。マクロ経済データでは、統計上、同一の商品に分類される財が輸出されると同時に輸入されるという現象がしばしば観察される。その結果として、自国の生産者が国内で生産する国産財が国内市場にも海外にも供給されると同時に、国内市場には国産財（自国で生産され国内市場に供給される財）に加えて同一の商品と見なされる輸入財も供給されることになる。

1) 完全代替性と完全変形可能性を仮定すると、貿易形態は一方向貿易になる。

もし国産財と外国財が全く同一の商品であるならば、小国の仮定の下ではこのような事態は生じないはずである。どちらか価格の低い財が国内市場を 100% 獲得して、他方を国内市場から完全に駆逐してしまうはずであるからである²⁾。

双方向貿易と呼ばれるこのような現象をモデルの中で取り扱うために、われわれは以下のような想定をおく。国内で（家計および政府の最終需要として、また企業の間需要として）需要される商品には、自国が生産し国内市場に供給するものと、海外で生産され輸入されて国内で販売されるものがある。両者は同じ財に分類される。われわれは、国内で販売される国産財と輸入財は互いに不完全代替財であると想定して、Armington (1969) の仮定を利用してモデルを定式化する。具体的には、国内の財市場に供給される商品 c は、国産財と輸入財から「生産」される合成商品であると考え、その生産技術は代替の弾力性一定 CES の集計関数³⁾によって表されるものとしてモデル化する。このとき、国産財と輸入財が不完全代替財になるように、Armington 関数の指数の値を設定する。

輸入される商品 $c \in CM$ （すなわち、輸入財を表す集合 CM に属する商品）について、その輸入量を QM_c 、国内で生産され国内で販売される数量を QD_c とすると、合成商品 c の国内供給量 QQ_c は、Armington 関数と呼ばれる次の CES 関数（1次同次）により与えられる。

$$QQ_c = \gamma_c^q (\delta_c^q QM_c^{-\rho_c^q} + (1 - \delta_c^q) QD_c^{-\rho_c^q})^{-\frac{1}{\rho_c^q}} \quad c \in CM$$

すなわち、Armington 関数は、国内供給される国産財と輸入財の代替関係を特定する。ただし、 γ_c^q は Armington 関数の規模係数、 δ_c^q は輸入財の加重、 ρ_c^q は係数である。なお、代替の弾力性 σ_c^q は、

-
- 2) この主張は厳密に言えば、外国財の方が安い場合には、このままで正しい。国産財の方が安い場合には、「国内の供給能力で国内需要を満たすことができる限り」という条件がつく。
 - 3) この取扱いの考案者にちなんで、この関数は Armington 関数と呼ばれる。

$$\sigma_c^q = \frac{1}{1 + \rho_c^q}$$

により与えられる。

必要とされる国内供給量 QQ_c が与えられたとき、国産財の国内供給量 QD_c と輸入財の供給量 QM_c は、与えられた国産財の国内供給価格 PD_c と輸入価格（自国通貨建て） PM_c の下で、この国内供給量 QQ_c を最小の費用で調達するように決定される。よって、輸入財と国産財の最適組合せを示す輸入財—国産財需要比率 $\frac{QM_c}{QD_c}$ は、国産財の国内供給価格 PD_c と輸入価格（自国通貨建て） PM_c の比率（相対価格）の関数として、

$$\frac{QM_c}{QD_c} = \left(\frac{PD_c}{PM_c} \frac{\delta_c^q}{1 - \delta_c^q} \right)^{\frac{1}{1 + \rho_c^q}} \quad c \in CM$$

により与えられる。

経済学的には、Armington 関数は需要者達の国産財と輸入財に対する選好が CES 関数として表されることを意味しており、無差別曲線が原点に対して凸になるように、 ρ_c^q の値が制限 ($1 < \rho_c^q < \infty$) されていると解釈することができる。国産財と輸入財の組み合わせは、(i) Armington 関数が特定する代替関係の技術的制約の下で、(ii) 国内供給の総調達費用を最小にするように組み合わせられると想定する。

ここで、輸入財と国産財の最適組合せの選択は、需要分類別ではなく商品別に行われていることに注意する必要がある。それは、国産財の国内需要と輸入財需要を需要分類別に分離するためのデータが存在しないためである。国産財の国内供給量 QD_c のデータは国民経済計算や産業連関表からは得られないので、モデル構築者が自分で算出するしかない。また、パラメーター値についても、Armington 関数の規模係数 γ_c^q と輸入財の加重 δ_c^q の値は基準均衡の社会勘定行列のデータから算出できるが、係数 ρ_c^q 、したがって代替の弾力性の値は社会勘定行列からは得られない。他の情報

を入手する必要がある。

なお、商品 c が非輸入財である場合 ($c \in CNM$, すなわち非輸入財を表す集合 CNM に属する場合) には、合成商品 c の供給量 QQ_c は国産財の国内供給量 QD_c に等しい。

$$QQ_c = QD_c \quad c \in CNM$$

次に、国内で生産される国産財が国内市場に供給されると同時に輸出され海外でも販売されるという現象のモデル化を検討しよう。このために、国内で生産される国産財は「変形」されて、国内市場への供給分と海外への供給分（輸出財）に作り替えられるものと考え、両者の間には不完全変形可能性があるとして仮定する。すなわち、商品 c の国内生産量 QX_c は輸出分 QE_c と国内供給分 QD_c とに分けられるが、その振り分けは変形関数と呼ばれる変形の弾力性一定の関数 CET を通じて行われると仮定する。

$$QX_c = \left(\gamma'_c \delta'_c QE_c^{\rho'_c} + (1 - \delta'_c) QD_c^{\rho'_c} \right)^{\frac{1}{\rho'_c}} \quad c \in CE$$

ただし、 γ'_c は変形関数の規模係数、 δ'_c は輸出財の加重、 ρ'_c は係数である。なお、変形の弾力性 σ'_c は、

$$\sigma'_c = \frac{1}{1 + \rho'_c}$$

により与えられる。このように、変形関数は国内供給分と輸出分との間の変形の弾力性を特定する。

商品 c の国内生産量 QX_c が与えられたとき、輸出量 QE_c と国内供給量 QD_c への振り分けは、与えられた商品 c の輸出価格（自国通貨建て） PE_c と国内供給価格 PD_c の下で、収入 $PD_c QX_c$ を最大にするように選択される。すなわち、輸出—国内供給比率 $\frac{QE_c}{QD_c}$ は、商品 c の輸出価格（自国通貨建て） PE_c と国内供給価格 PD_c の比率（相対価格）の関数

$$\frac{QE_c}{QD_c} = \left(\frac{PE_c}{PD_c} \frac{1 - \delta_c^t}{\delta_c^t} \right)^{\frac{1}{\rho_c - 1}}$$

として与えられる。この式と上の輸入財—国産財需要比率との違いは、輸入量 QM_c は輸入価格 PM_c の減少関数であるのに対して、輸出量 QE_c は輸出価格 PE_c の増加関数であることである。

Armington 関数と同様に、変形関数の規模係数と輸出財の加重の値は基準均衡の社会勘定行列のデータから算出できるが、係数 ρ_c 、したがって変形の弾力性 σ_c の値は社会勘定行列からは得られない。他の情報が必要である。経済学的には、この変形関数の等量線が原点に向かって凹になるように、 ρ_c の値が制限 $(1 - \rho_c < \infty)$ されると解釈することができる。

なお、商品 c が非輸出財である場合 ($c \in CNE$ 、すなわち非輸出財を表す集合 CNE に属す場合) には、商品 c の国内生産量 QX_c は全て国内市場に供給され輸出されないので、国内需要量 QD_c は国産財の国内供給量 QX_c と等しい。

$$QX_c = QD_c \quad c \in CNE$$

最後に、表1から分かるように、全ての商品について、

$$QX_c + QM_c = QQ_c + QE_c \quad c \in C$$

という関係が成り立つ。しかし、

表1 国内生産，輸出，輸入の関係

		需 要 者		
		国内	国外	計
供給者	国内	QD_c	QE_c	QX_c
	国外	QM_c		QM_c
	計	QQ_c	QE_c	

$$QE_c = QM_c$$

でない限り、その他の数量の間にバランス関係はないことに注意しておく。

3. モデルの方程式体系

モデルを表す方程式体系（全部で27式）を、価格、生産および商品、経済主体、体系制約の4ブロックに分割して考えよう。本稿においても、GAMS のプログラミングを簡潔にするために、変数の添え字を表す以下の集合を用いてモデルを構築する。

$a \in A = \{FA, FB\}$	企業 = {企業 <i>FA</i> , 企業 <i>FB</i> }
$c \in C = \{CA, CB\}$	商品 = {商品 <i>CA</i> , 商品 <i>CB</i> }
$c \in CM (\subset C)$	輸入財
$c \in CNM (\subset C)$	非輸入財
$c \in CE (\subset C)$	輸出財
$c \in CNE (\subset C)$	非輸出財
$f \in F = \{K, L\}$	生産要素 = {資本 <i>K</i> , 労働 <i>L</i> }
$h \in H = \{HA, HB\}$	家計 = {家計 <i>HA</i> , 家計 <i>HB</i> }
$i \in I = \{HA, HB, GOV, RoW\}$	経済主体 = {家計 <i>HA</i> , 家計 <i>HB</i> , 政府 <i>GOV</i> , 海外部門 <i>RoW</i> }

前稿（小平 (2002d)）の age 4 モデルと比較して、ここで新たに追加される集合は、輸入財の集合 *CM*、非輸入財の集合 *CNM*、輸出財の集合 *CE*、非輸出財の集合 *CNE* の4つであり、何れも商品全体を表す集合 *C* の部分集合である。また、経済主体を表す集合 *I* には、新たな成分として海外部門 *RoW* が追加される。

本稿でも、記号の使い方について、小平 (2002a) の約束を引き続き維持

して、(i) 内生変数を表すには大文字を使い、パラメーター（固定値あるいは外生的な値をもつ変数を含む）を表すには小文字を使う、また (ii) 数量を表す変数あるいはパラメーターを Q あるいは q で始める、商品価格を表す場合には P あるいは p で始める、要素価格を表す場合には W あるいは w で始めることとする。パラメーターは以下の通りである。

cpi = 消費者価格指数 CPI

$cwts_c$ = 消費者価格指数 CPI における商品 c の加重

ica_{ca} = 企業 a の産出 1 単位当たりの商品 c の中間投入量

pwe_c = 商品 c の国際価格（外貨建て輸出価格）

pwm_c = 商品 c の国際価格（外貨建て輸入価格）

qg_c = 商品 c の政府需要

$\overline{qinv_c}$ = 商品 c の基準年投資需要量

$shry_{hf}$ = 要素 f の総要素所得のうち家計 h への分配率

te_c = 商品 c の輸出関税率（マイナスならば補助金率）

tm_c = 商品 c の輸入関税率

tq_c = 商品 c の消費税率

$tr_{ii'}$ = 経済主体 i' から経済主体 i への移転

ty_h = 家計 h の所得税率

α_{hc} = 家計 h の効用関数における商品 c の係数（商品 c への支出割合）

β_{fa} = 企業 a の生産関数における要素 f の係数（要素 f への分配率）

λ_a = 企業 a の生産関数の規模係数（効率性パラメーター）

θ_{ac} = 企業 a の活動水準 1 単位当たりの商品 c の産出量

γ_c^q = 合成商品 c の Armington 関数の規模係数

δ_c^q = 合成商品 c の Armington 関数における輸入財の加重

ρ_c^q = 合成商品 c の Armington 関数の係数

σ_c^q = 合成商品 c における国産財と輸入財の間の代替の弾力性

γ_c^t = 合成商品 c の輸出財変形関数の規模係数

δ'_c = 合成商品 c の輸出財変形関数における輸出財の加重

ρ'_c = 合成商品 c の輸出財変形関数の係数

σ'_c = 合成商品 c における国内供給分と輸出分の間の変形の弾力性

ここで、商品 c の外貨建て輸出価格 pwe_c と輸入価格 pwm_c が、外生的に与えられるパラメーターとされていることに注意しよう。これは、自国は世界市場に比べて取引規模が小さく、国際価格に対して価格受容者として行動する国としてモデル化されていることを意味する。すなわち、本稿においては自国は「小国」と仮定されている。また、商品 c の輸出関税率 te_c 、輸入関税率 tm_c も新たなパラメーターとされており、これらの関税率はモデルの外部から政策的に与えられるものと見なされている。さらにこのモデルでは、Armington 関数と輸出財変形関数に関するパラメーターが新たに追加されている。

(i) 価格ブロック

双方向交易を伴う外国貿易の導入によって、価格体系はかなり複雑になる。このブロックの6本の方程式のうち、式(1)-(4)は新規の式である。

小国の仮定をおいたので、商品 c の自国通貨建て輸入価格 PM_c は、国際価格（外貨建て輸入価格） pwm_c が与えられると、関税調整 $(1 + tm_c) \times$ 為替レート EXR との積によって決定される。

$$(1) \quad PM_c = (1 + tm_c)EXR pwm_c \quad c \in CM$$

ここに、 tm_c は輸入関税率（マイナスならば、補助金率）、為替レート EXR は外貨1単位当たりの国内通貨量である。自国通貨建て輸出価格 PE_c も同様に、国際価格（外貨建て輸出価格） pwe_c が与えられると、関税調整 $(1 - te_c) \times$ 為替レート EXR との積によって決定される。

$$(2) \quad PE_c = (1 - te_c)EXR pwe_c \quad c \in CE$$

ここに、 te_c は輸出関税率である。なお、式 (1) と (2) はそれぞれ輸入される商品、輸出される商品にのみ当てはまる。

双方向貿易という現象をモデル化するために、第2節で検討したように、商品 c を需要する国内の経済主体（家計、生産者、投資家、政府）にとって、自国で生産され国内市場に供給される国産財と海外で生産され輸入されて国内で販売される輸入財の間には不完全な代替関係があると想定する。具体的には、国内供給量 QQ_c は、Armington 関数により記述される生産技術を用いて、国内供給される国産財 QD_c と輸入財 QM_c から「生産」されると想定する。このとき、商品 c の国内需要者価格 PQ_c における国内総支出額 $PQ_c QQ_c$ 、すなわちアブソープションは、国内で販売される国産財および輸入財への支出の総和（消費税を含む）に等しい。

$$(3) \quad PQ_c QQ_c = \left(PD_c QD_c + PM_c QM_c \Big|_{c \in CM} \right) (1 + tq_c)$$

$$c \in C$$

ここで、 PD_c は商品 c の国内供給価格であり、 QD_c は国内で生産された商品 c の国内供給量 QD_c の積であるから、右辺の () の中の第1項 $PD_c QD_c$ は消費税を課する前の自国需要者達の国産財への支出額を表している。また PM_c は輸入財 c の国内価格（自国通貨建て）であり、 QM_c はその輸入量であるから、第2項 $PM_c QM_c$ は輸入財への支出額（消費税を含まない）を表している。したがって、右辺全体はこれらの和に消費税を加えた総額を表しているから、国内で販売される国産財および輸入財への支出額（消費税込み）になる。式 (3) は、この支出額が、国産財と輸入財から「生産」される合成商品 c への支出額に等しいというアブソープションの定義を示している。この式の両辺を QQ_c で除することによって、合成商品 c の国内需要者価格 PQ_c を求めることができる。

国内で生産される商品は国内市場と海外に供給される。国内市場向けと輸出向けの間不完全変形可能性は、総生産量を2つの用途に変形すると

想定して、以下の式 (14) のように定式化される。このとき、商品 c の生産者価格で評価した消費税を除く国内生産額（すなわち、生産者価格 PX_c と国内生産量 QX_c の積）は、国内で生産された商品 c の国内への販売額と（自国通貨建ての）輸出額の和に等しい。

$$(4) \quad PX_c QX_c = PD_c QD_c + PE_c QE_c \Big|_{c \in CE} \quad c \in C$$

ここで、 PD_c は国内で生産された商品 c の国内供給価格であり、 QD_c はその国内供給量であるから、この式の右辺第 1 項は国内供給額を表しており、また PE_c は商品 c の自国通貨建て輸出価格、 QE_c は輸出量であるから、第 2 項は輸出額を表している。式 (4) の両辺を国内生産量 QX_c で除することによって、生産者価格 PX_c を求めることができる。

生産活動 a の価格は前稿と同様に、商品 c の（消費税を除く）生産者価格 PX_c と企業 a の活動水準 1 単位当たりの商品 c の産出量の積によって与えられる。

$$(5) \quad PA_a = \sum_{c \in C} PX_c \theta_{ac} \quad a \in A$$

そして、生産活動 a の付加価値（あるいは純）価格 PVA_a は、この生産活動価格から生産活動 1 単位当たりの投入費用を差し引けば得られる。

$$(6) \quad PVA_a = PA_a - \sum_{c \in C} PQ_c ica_{ca} \quad a \in A$$

ただし、 PQ_c は合成商品 c の国内供給価格であり、前稿とは記号が異なる。

(ii) 生産および商品ブロック

このブロックの方程式のうち、式 (7)-(10) は（一部の記号の変更を除けば）前稿の age 4 モデルと同じである。新たに追加された (11)-(16) は、国内で生産される商品 c の国内供給と輸出への振り分け方、および国内市場

に供給される合成商品 c を「生産」する国産財と輸入財の組み合わせの決定を説明する。非輸出財や非輸入財には、もっと簡単な式が当てはまる。

企業 a の生産関数には、これまでと同様に、Cobb-Douglas 型の生産技術を想定する。

$$(7) \quad QA_a = \lambda_a \prod_{f \in F} QF_{fa}^{\beta_{fa}} \quad a \in A$$

企業 a が雇用する要素 f に対して支払う要素価格は、要素価格の歪みを考慮すると経済全体の平均要素価格 WF_f と歪み係数 $WFDIST_{fa}$ の積として与えられるから、要素 f を QF_{fa} 単位雇用する企業 a が要素 f に支払う報酬総額は $WFDIST_{fa} WF_f QF_{fa}$ と書ける。ここでは Cobb-Douglas 型生産関数 (7) を想定しているので、生産要素 f への要素報酬は生産額 $PVA_a QA_a$ の一定割合に等しくなる。したがって、生産要素 f の要素(逆)需要関数は、

$$(8) \quad WFDIST_{fa} WF_f = \frac{\beta_{fa} PVA_a QA_a}{QF_{fa}} \quad f \in F, a \in A$$

により与えられる。

引き続き、固定係数 (Leontief 技術) の中間投入を想定して、中間投入係数を ica_{ca} とすると、商品 c に対する企業 FA の中間需要 $QINT_{ca}$ は、

$$(9) \quad QINT_{ca} = ica_{ca} QA_a \quad c \in C, a \in A$$

と表される。商品 c の市場生産量は、両企業の商品 c の生産量の和となる。

$$(10) \quad Q_c = \sum_{a \in A} \theta_{ac} QA_a \quad c \in C,$$

ただし、結合生産を想定していないので、企業 a の活動水準 1 単位当たりの商品 c の産出量を表すパラメーター θ_{ac} の値は 0 または 1 である。上述したように、式 (7)–(10) には記号以外の age 4 モデルからの変更は

ない。

国内需要者達は国内で生産され国内市場に供給される商品も、海外で生産され輸入される商品も購入するが、両者は互いに不完全代替財であると考えている。ここでは、合成商品 c が国産財と輸入財から「生産」され国内に供給されると考えて、代替の弾力性一定 CES の Armington 集計関数

$$(11) \quad QQ_c = \gamma_c^q \left(\delta_c^q QM_c^{-\rho_c^q} + (1 - \delta_c^q) QD_c^{-\rho_c^q} \right)^{\frac{1}{\rho_c^q}} \quad c \in C$$

を使ってモデル化する。ただし、この式の定義域は CM に属する商品に限られる。Armington 関数 (11) の意味づけについては、第 2 節で説明したのでここでは繰り返さない。

式 (11) だけでは、輸入財と国産財の最適な組合せは決まらない。最適な輸入財—国産財需要比率 $\frac{QM_c}{QD_c}$ は、国産財と輸入財の価格が与えられたとき、Armington 関数により示される「生産」技術の制約の下で合成商品の供給費用 $PD_c QD_c + PM_c QM_c$ を最小にする問題の解として決まり、国産財の国内供給価格 PD_c と輸入価格（自国通貨建て） PM_c の比率の関数として表される。

$$(12) \quad \frac{QM_c}{QD_c} = \left(\frac{PD_c}{PM_c} \frac{\delta_c^q}{1 - \delta_c^q} \right)^{\frac{1}{1+\rho_c^q}} \quad c \in CM$$

なお、商品 c が非輸入財である場合には、合成商品 c の供給量 QQ_c は国産財の国内供給量に等しいので、(11) (12) に代わり、

$$(13) \quad QQ_c = QD_c \quad c \in CNM$$

が成立する。

次に、国内で生産される商品は国内市場に供給されると同時に輸出されている。両者の間の不完全変形可能性は、すでに第 2 節で検討したように、

国内市場に供給される国産財と輸入財の間の不完全代替可能性と同様に扱われる。

$$(14) \quad QX_c = \left(\gamma'_c \delta'_c QE_c^{\rho'_c} + (1 - \delta'_c) QD_c^{\rho'_c} \right)^{\frac{1}{\rho'_c}} \quad c \in CE$$

ここで、変形関数 (14) は変形の弾力性一定の関数 CET として定式化されるが、変形の弾力性 σ'_c が負になること以外は、Armington 関数 (CES 関数) と同一である。

式 (14) だけでは輸出量と国内供給量の振り分けは決まらない。最適な輸出—国内供給比率 $\frac{QE_c}{QD_c}$ は、与えられた国内生産量 QX_c を国内供給 QD_c と輸出 QE_c とに振り分ける変形関数 (14) の制約の下で、国内需要価格 PD_c と輸出価格 PE_c が与えられたとき、企業の収入最大化問題の解として、商品 c の輸出価格 (自国通貨建て) PE_c と国内供給価格 PD_c の比率の関数

$$(15) \quad \frac{QE_c}{QD_c} = \left(\frac{PE_c}{PD_c} \frac{1 - \delta'_c}{\delta'_c} \right)^{\frac{1}{\rho'_c - 1}} \quad c \in CE$$

として与えられる。これは輸出供給を与える方程式と見なせる。式 (12) と (15) の違いは、輸入量 QM_c は輸入価格 PM_c の減少関数であるのに対して、輸出量 QE_c は輸出価格 PE_c の増加関数であることである。

なお、商品 c が非輸出財である場合には、商品 c の国内生産量 QX_c と国産財の国内供給量 QD_c は等しいので、(14) (15) に代わり、

$$(16) \quad QX_c = QD_c \quad c \in CNE$$

が成立する。

(iii) 経済主体ブロック

このブロックにおける方程式の変更点は、海外部門との取引 (輸出, 輸

入と移転)が家計の収入，政府の収入と支出の定義に登場することである。それら以外は，age 4 モデルと変わらない。

企業 a が支払う要素価格は $WFDIST_{fa} WF_f$ であるから，この企業の要素 f の雇用量を QF_{fa} とすると，要素 f の総要素所得は $\sum_{a \in A} WFDIST_{fa} WF_f QF_{fa}$ により与えられる。このうち家計 h が受け取る割合は $shry_{hf}$ であるから，家計 h が要素 f から受け取る要素所得 YF_{hf} は，

$$(17) \quad YF_{hf} = shry_{hf} \sum_{a \in A} WF_f WFDIST_{fa} QF_{fa} \quad h \in H, f \in F$$

となる。

次に，家計 h の総所得 YH_h は，自己が保有する資本と労働に対する要素所得，国内政府からの移転 $tr_{h,gov}$ ，海外部門からの移転 $EXR tr_{h,row}$ の和により与えられる。

$$(18) \quad YH_h = \sum_{f \in F} YF_{hf} + tr_{h,gov} + EXR tr_{h,row} \quad h \in H$$

式 (18) の左辺第 3 項は，為替レート EXR と海外部門 RoW からこの家計 h への移転 $tr_{h,row}$ (外貨建て) の積であり，自国通貨建ての海外部門からの移転額を示している。

式 (18) の家計所得 YH_h には，税率 ty_h の所得税が課税されるので，可処分所得は $(1 - ty_h) YH_h$ となる。家計 h はこの可処分所得の一定割合 MPS_h を貯蓄し，一定割合 $(1 - MPS_h)$ を今期の消費支出に充てる。すなわち，総消費支出額は $(1 - MPS_h)(1 - ty_h) YH_h$ となる。さらにここでは，Cobb-Douglas 型の効用関数を想定するので，家計は総消費支出額の一定割合 α_{ch} を商品 c の購入に支出する。したがって，家計 h の商品 c に対する (逆) 需要関数は，

$$(19) \quad QH_{ch} = \frac{\alpha_{ch}(1 - MPS_h)(1 - ty_h) YH_h}{P_c} \quad c \in C, h \in H$$

により与えられる。

家計部門の貯蓄により投資が行われる。投資方程式は、

$$(20) \quad QINV_c = IADJ \overline{qinv}_c \quad c \in C$$

により与えられる。ただし、 \overline{qinv}_c は商品 c の基準時点投資需要量、 $IADJ$ は比例的調整係数である。

最後に、政府行動を特定する。政府収入 YG は、所得税、消費税、輸入関税、輸出関税からの税收、海外部門から政府への移転から成る。

$$(21) \quad YG = \sum_{h \in H} ty_h YH_h + \sum_{c \in C} tq_c \left(PD_c QD_c + PM_c QM_c \Big|_{c \in CM} \right) \\ + \sum_{c \in CM} tm_c EXRpwm_c QM_c + \sum_{c \in CE} te_c EXRpwe_c QE_c \\ + EXRtr_{gov, row}$$

ここで、所得税は家計所得 YH_h に対して税率 ty_h で課税されるので、右辺第1項の $\sum_{h \in H} ty_h YH_h$ は所得税収を表している。消費税は国産財の国内供給額 $PD_c QD_c$ と輸入財取引額 $PM_c QM_c$ (自国通貨建て) に対して税率 tq_c で課税されるので、第2項 $\sum_{c \in C} tq_c \left(PD_c QD_c + PM_c QM_c \Big|_{c \in CM} \right)$ は消費税収である。為替レート EXR と輸入財の国際価格 pwm_c の積は自国通貨建ての輸入価格を表すので、輸入量 QM_c にこれを掛けた積 $EXRpwm_c QM_c$ は自国通貨建て輸入額を意味する。 tm_c は輸入関税率であるから、第3項の $\sum_{c \in C} EXRpwm_c QM_c$ は輸入関税収に等しい。同様に、第4項の $\sum_{c \in CE} EXRpwe_c QE_c$ は輸出関税収を示す。第5項は、海外部門 RoW から政府 GOV への移転 $tr_{gov, row}$ (外貨建て) を自国通貨建てに換算した金額になる。

政府支出 EG は、政府の商品需要 qg_c への支出 $PQ_c qg_c$ と政府 gov から家計 h への移転 $tr_{h, gov}$ から成る。

$$(22) \quad EG = \sum_{c \in C} PQ_c qg_c + \sum_{h \in H} tr_{h, gov}$$

ここで、右辺第1項は政府の商品 c に対する支出 $PQqg_c$ を全ての商品について合計したもの、すなわち商品需要総額である。このモデルでは政府も消費税を負担すると想定しているので、消費税を含まない生産者価格 PX_c ではなく国内供給価格 PQ_c が適用される。第2項は家計への移転 $tr_{h,gov}$ を全ての家計について合計したもの、すなわち移転総額である。

(iv) 体系制約ブロック

方程式体系最後のこのブロックは、このモデルが全体として満たす制約を定義する。前稿のモデルに比べて、海外部門制約 (式 (25)) はここで新しく加えられており、商品市場と貯蓄—投資のバランス条件 (式 (24) と (26)) には修正が施されている。要素市場の取扱い (式 (23)) には変更はない。全ての商品と要素について伸縮的な価格が市場を清算すると仮定する。政府については貯蓄が貯蓄—投資バランスを清算するのに対して、投資額は総貯蓄額の変化に合わせて調整されると想定し、海外部門については伸縮的な為替レートか伸縮的な海外部門貯蓄の何れかを想定する。

要素市場のバランス条件は、

$$(23) \quad \sum_{a \in A} QF_{fa} = QFS_f \quad f \in F$$

により与えられる。すなわち、要素 f の需要は要素 f の供給に等しい。

合成商品の市場バランス条件は、次により与えられる。

$$(24) \quad QQ_c = \sum_{a \in A} QINT_{ca} + \sum_{h \in H} QH_{ch} + qg_c + QINV_c \quad c \in C$$

上式の左辺は、国内供給される国産財と輸入財を集計する Armington 関数 (11) により与えられる合成商品 c の国内供給 QQ_c を示している (なお、非輸入財については、国産財の国内供給量に等しい。(13) 参照)。 $QINT_{ca}$ は生産活動 a における商品 c の中間投入量、 QH_{ch} は家計 h による商品 c の消費、 qg_c は政府による商品 c の需要、 $QINV_c$ は商品 c に対する投資

需要量であるから、右辺は合成商品 c の中間需要+家計需要+政府需要+投資需要の和を表している。この市場がバランスするように、合成商品 c の国内供給価格 PQ_c が決定される。なお、外国貿易が行われない場合 (age 1 モデルから age 4 モデルまで) には、商品市場は国内生産量 (=供給量) と国内需要量が等しいときにバランスしていた。

合成商品 c の国内供給 QQ_c 以外に、このモデルには以下の数量を表す変数がある。

QA_a = 生産活動 a の水準

QX_c = 商品 c の国内生産量

QD_c = 国内で生産された商品 c の国内供給量

QE_c = 商品 c の輸出量

QM_c = 商品 c の輸入量

これらの変数は供給量と需要量の両方を表している (すなわち、供給量と需要量として均衡値が代入されている)。輸出量 QE_c と輸入量 QM_c については、需要量と供給量とその市場をバランスさせる⁴⁾。この他の数量変数については、その価格変数 (生産活動 a の価格 PA_a , 商品 c の (消費税を除く) 生産者価格 PX_c , 国内で生産された商品 c の国内供給価格 PD_c) が市場をバランスさせる。

海外部門との経常収支バランス (外貨建て) は、次の式により与えられる。

$$(25) \quad \sum_{c \in C} pwe_c QE_c + \sum_{i \in I} tr_{i, row} + FSAV = \sum_{c \in CM} pwm_c QM_c$$

ここで、左辺は外貨の稼得を表しており、輸出受け取り、海外部門からの自国経済主体 (家計と政府) への移転、海外部門貯蓄 $FSAV$ (何れも外貨建て) から成る。海外部門貯蓄は経常収支赤字に等しい。一方、外貨の支払いを表す右辺は輸入代金 (外貨建て) の和から成る。経常収支をバランスさせる役割を果たす変数は、為替レート EXR と海外部門貯蓄 $FSAV$ の 2

4) 小国の仮定の下では、世界市場の需要と供給の弾力性は無限大になる。

つあるが、以下では海外部門貯蓄 $FSAV$ は固定されていると仮定する。

資本市場のバランス条件は

$$(26) \quad \sum_{c \in C} PQ_c QINV_c + WALRAS \\ = \sum_{h \in H} MPS_h (1 - ty_h) YH_h + (YG - EG) + EXRFSAV$$

により与えられる。式 (26) の左辺の資本市場の需要側（投資側）は age 4 モデルと同じであり，投資支出+WALRAS ダミー変数⁵⁾を表している。右辺の供給側（貯蓄側）には自国通貨建てに変換された海外部門貯蓄が追加され，家計貯蓄+政府貯蓄+海外部門貯蓄を表している。為替レート EXR と海外部門貯蓄 $FSAV$ のどちらか1つが固定されている限り，貯蓄—投資の閉じ方は影響されない。

最後に，価格正規化方程式はこれまでと同じである。

$$(27) \quad \sum_{c \in C} cwtsc PQ_c = cpi$$

4. 基準均衡データセット

基準均衡として，次のような社会勘定行列（表2）で与えられる数値例⁶⁾を考えよう。また仮設均衡として，為替レートは伸縮的であると仮定して，商品 CA の（外貨建て）国際価格が25%上昇する場合を検討する。

応用一般均衡分析の慣習に倣い，全ての価格について基準均衡値を1とする。ただし，前稿（小平 (2002d)）で指摘したように，中間投入および消費税の存在により，基準均衡値を1と仮定できないものがある。すなわち，中間投入のために，企業 FA の企業価格 PVA_a の初期値は1以外の値になる。また，ここでは生産者価格 PX_c の初期値を1と想定するので，商

5) $WALRAS$ は一般均衡において0となるダミー変数である。

6) この数値例は，Lofgren (1999) による。

表2 基準均衡の社会勘定行列

	企業		商品		資本	労働	家計		政府	貯蓄-投資	所得税	売上税	関税	海外部門	合計
	FA	FB	CA	CB	K	L	HA	HB	GOV	S-I	YTAX	STAX	TAR	ROW	
企業	279														279
FB		394													394
商品	84	55					30	49	13	28				30	289
CB	50	99					165	92	67	85					558
資本	73	135													208
K	72	105													177
労働															
L															
家計					125	95			25					40	285
HA					83	82			5					16	186
HB											25	30	39	15	109
政府															
GOV															
S-I															
貯蓄-投資															
YTAX							70	40	-1					4	113
所得税															
STAX															
消費税	10						20		5						25
STAX															30
TAR															39
関税															
ROW															
海外部門															
ROW															
合計	279	394	289	558	208	177	285	186	109	113	25	30	39	105	105

品 c の消費者価格 P_c の初期値は $1 + \text{消費税率}$ となる。

企業価格 PVA_a は初期生産活動価格 PA_a と社会勘定行列 SAM の生産活動列のデータよりカラブレイトされる。消費税率を求めれば、消費者価格 P_c は算出できる。想定するモデル類型に応じて、要素価格の歪みを考慮する。

貿易の導入は社会勘定行列をどのように変えるであろうか。社会勘定行列は拡張されて、海外部門 RoW (新たな経済主体) と輸入関税 TAR (新たな税収項目) という2つの勘定が追加される。海外部門 RoW 勘定の行を横に見ていくと、海外部門 RoW から自国に商品 CB が供給されている(つまり、自国は商品 CB を海外部門 RoW から輸入している。輸入額は105) ことが分かる。また、海外部門勘定の行は、輸入財に対する私たちの支出が自国との取引における海外部門の唯一の収入であることを

示している。関税 TAR の行を横に見ていくと、自国の政府はこの取引に関わる関税 39 を受け取っている（すなわち、輸入関税 39 が政府に納められている）ことが分かる。

一方、海外部門 RoW 勘定の列は、海外部門 RoW からの自国の受け取りは、輸出収入と自国の諸家計および自国政府に対する移転からなることを示しているから、この列を縦に見ていくと、海外部門は自国から商品 CA を購入している（つまり、自国は商品 CA を海外部門に輸出している。輸出額は 30）こと、海外部門 RoW から家計 HA は 40 の、家計 HB は 16 の、自国の政府 GOV は 15 の移転を受けていることが分かる。海外部門 RoW から貯蓄—投資勘定 $S-I$ への支払いは、海外部門貯蓄あるいは経常勘定赤字、すなわち自国の（資本勘定ではなく）経常勘定の外貨支出と収入の差を表しており、この数値例では 4 である。

両生産要素の要素市場の取扱いは、前稿（小平 (2002d)）の age 4 モデルと同じである。要素賦存量と労働雇用についてのデータは変わっていない。商品 CB の輸入財と国内供給の間の代替の弾力性の値は 0.7、商品 CA の輸出と国内供給の間の変形の弾力性の値は 2 と想定されている。

5. GAMS 入力ファイル

上の第 2 節および第 3 節で検討したモデル体系と、第 4 節の基準均衡データセットに基づく一般均衡体系を、数理計画法解法プログラム GAMS のプログラムとして表現したものが、以下の age 5 モデルの入力ファイル（表 3）である。表の左側の数字は説明のために付けた行番号であり、GAMS 入力には無関係である。age 4 モデルと異なる箇所には、行番号に *（アスタリスク）を付した。この入力ファイルでは、最初にモデル・パラメーターのカラブレイション、続いて基準均衡の再現テストを行い、最後に為替レートは伸縮的であると仮定して商品 CA の（外貨建て）世界価格が 25% 上昇した場合の仮設均衡を求めている。

GAMS による応用一般均衡分析：海外部門のあるモデル

表3 age 5 モデルの入力ファイル

```

*1 $TITLE AGES5
2 $OFFUPPER
3 *$ONSYMKREF
4
5 *INTRODUCTION=====
6 $ONTEXT
7
8 age5a = Open-economy model with Rest of World
9       Foreign Trade = CET-Armington specification
10      (1) S-I Balance = investment-driven savings
11      (2) Capital Market = firm-specific and fully employed
12      (3) Labor Market = mobile and unemployed (fixed wages)
13      (4) Foreign Exchange Market =
14          flexible exchange rate & fixed foreign savings
15
16 $OFFTEXT
17
18 *SETS=====
19
20 SETS
21
22 AC global set (SAM accounts and other items)
23   /FA   firm fa
24   FB   firm fb
25   CA   commodity ca
26   CB   commodity cb
27   K   capital
28   L   labor
29   HA   household ha
30   HB   household hb
31   GOV  government
32   S-I  savings-investment
33   YTAX income tax
34   STAX sales tax
*35   TAR  import tariff
*36   ROW  rest of world
37   TOTAL total account in SAM /
38
39 ACNT(AC) all elements in AC except total
40
41 A(AC) activities
42   /FA, FB/
43
44 C(AC) commodities
45   /CA, CB/
46

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

*47 CE(C)  exported commodities
*48       /CA/
49
*50 CNE(C) non-exported commodities
*51       /CB/
52
*53 CM(C)  imported commodities
*54       /CB/
55
*56 CNM(C) non-imported commodities
*57       /CA/
58
59 F(AC)  factors
60       /K, L/
61
62 I(AC)  institutions
63       /HA, HB, GOV, ROW/
64
65 H(I)   households
66       /HA, HB/
67       ;
68
69 ALIAS(AC,ACP); ALIAS(C,CP); ALIAS(F,FP); ALIAS(I,IP);
70 ACNT(AC) = YES; ACNT('TOTAL') = NO; ALIAS(ACNT,ACNTP);
71
72
73 *PARAMETERS=====
74
75 PARAMETERS
76
77 cpi           consumer price index
78 cwts(C)      weight of commodity c in the CPI
79 ica(C,A)     firm a's intermediate input of commodity c
*80 pwe(C)      world price for export c (foreign currency)
*81 pwm(C)      world price for import c (foreign currency)
82 qg(C)        government demand for commodity c
83 qinvbar(C)   base-year investment demand for commodity c
84 shry(H,F)    household h's share in factor f's income
*85 te(C)       export subsidy rate for commodity c
*86 tm(C)       import tariff rate for commodity c
87 tq(C)        sales tax rate on commodity c
88 tr(I,IP)     transfer from institution ip to i
89 ty(H)        income tax rate on household h
90
91 alpha(C,H)   commodity c's share in household h's spending
92 beta(F,A)    factor f's share in firm a's value-added
*93 gammaq(C)   Armington function shift parameter for commodity c
*94 gammat(C)   CET function shift parameter for commodity c

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

*95 deltag(C)   Armington function share parameter for commodity c
*96 deltat(C)  CET function share parameter for commodity c
  97 lambda(A)  scale parameter for firm a's production
*98 rhoq(C)    Armington function exponent for commodity c
*99 rhot(C)    CET function exponent for commodity c
100 theta(A,C) commodity c's output per unit of production by firm a
101 ;
102
103 *VARIABLES=====
104
105 VARIABLES
106
107 EG          government expenditures
*108 EXR       exchange rate (dom. currency per unit of for. currency)
*109 FSAV      foreign savings (foreign currency)
110 IADJ        investment adjustment coefficient
111 MPS(H)      mps (=aps) for household h
112 PA(A)       price of firm a
*113 PD(C)     domestic price of domestic output c
*114 PE(C)     export price for c (domestic currency)
*115 PM(C)     import price for c (domestic currency)
*116 PQ(C)     composite commodity price for c
117 PVA(A)     value-added (or net) price for firm a
118 PX(C)       producer price of commodity c
119 QA(A)       production level of firm a
*120 QD(C)     quantity sold domestically of domestic output c
*121 QE(C)     quantity of exports for commodity c
122 QF(F,A)    firm a's demand of factor f
123 QFS(F)     supply of factor f
124 QH(C,H)    household h's demand of commodity c
125 QINT(C,A)  firm a's intermediate demand of commodity c
126 QINV(C)    investment demand for commodity c
*127 QM(C)     quantity of imports of commodity c
*128 QQ(C)     quantity of goods supplied domestically (composite supply)
*129 QX(C)     quantity of domestic output of commodity c
130 WALRAS     dummy variable (zero at equilibrium)
131 WF(F)      average price of factor f
132 WFDIST(F,A) distortion factor for factor f in firm a
133 YF(H,F)    household h's income from factor f
134 YG         government revenue
135 YH(H)      household h's income
136 ;
137
138 *EQUATIONS=====
139
140 EQUATIONS
141
*142 *PRICE BLOCK+++++

```


GAMS による応用一般均衡分析：海外部門のあるモデル

```

*143 PEDEF(C)      export price for commodity c (domestic currency)
*144 PMDEF(C)      import price for commodity c (domestic currency)
*145 ABSORB(C)     absorption for commodity c
*146 OUTVAL(C)     output value for commodity c
147 PADEF(A)       price for firm a
148 PVADEF(A)      value-added price for firm a
149
150 *PRODUCTION AND COMMODITY BLOCK+++++++
151 PRODFN(A)       firm a's production function
152 FACDEM(F,A)     firm a's demand for factor f
153 INTDEM(C,A)     firm a's intermediate demand for commodity c
154 OUTPUTFN(C)     output of commodity c
*155 ARM(C)         composite supply for commodity c
*156 ARMM(C)        composite supply for non-imported commodity c
*157 CET(C)         output transformation for commodity c
*158 CETNE(C)       output transformation for non-exported commodity c
*159 EDRATIO(C)    export-domestic supply ratio for commodity c
*160 IDRATIO(C)    import-domestic demand ratio for commodity c
161
162 *INSTITUTION BLOCK+++++++
163 FACTRNS(H,F)    transfer from factor f to household h
164 GOVREV          government revenue
165 GOVEXP          government expenditures
166 HHDINC(H)       income of household h
167 HHDEM(C,H)     household h's demand of commodity c
168 INVDEM(C)       investment demand for commodity c
169
170 *SYSTEM CONSTRAINT BLOCK+++++++
171 FACTEQ(F)       market balance for factor f
172 COMEQ(C)        market balance for commodity c
*173 CURACC         current account balance for RoW
174 SAVINV          savings-investment balance
175 PNORM           price normalization
176 ;
177
178 *PRICE BLOCK+++++++
179
*180 PEDEF(C)$CE(C).. PE(C) =E= (1 - te(C))*EXR*pwe(C);
181
*182 PMDEF(C)$CM(C).. PM(C) =E= (1 + tm(C))*EXR*pwm(C);
183
*184 ABSORB(C)..    PQ(C)*QQ(C) =E=
*185                (PD(C)*QD(C)
*186                + (PM(C)*QM(C))$CM(C))*(1 + tq(C));
187
*188 OUTVAL(C)..    PX(C)*QX(C)
*189                =E= PD(C)*QD(C) + (PE(C)*QE(C))$CE(C);
190

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

191 PADEF(A)..          PA(A) =E= SUM(C, PX(C)*theta('C));
192
193 PVADEF(A)..         PVA(A) =E= PA(A) - SUM(C, PQ(C)*ica(C,A));
194
195
196 *PRODUCTION AND COMMODITY BLOCK+++++++
197
198 PRODFN(A)..         QA(A) =E=
199                      lambda(A)*PROD(F, QF(F,A)**beta(F,A));
200
201 FACDEM(F,A)..       WF(F)*WFDIST(F,A) =E=
202                      beta(F,A)*PVA(A)*QA(A) / QF(F,A);
203
204 INTDEM(C,A)..       QINT(C,A) =E= ica(C,A)*QA(A);
205
206 OUTPUTFN(C)..       QX(C) =E= SUM(A, theta(A,C)*QA(A));
207
*208 ARM(C)$CM(C)..     QQ(C) =E= gammaq(C)*
*209                      (deltaq(C)*QM(C)**(-rhoq(C)) + (1-deltaq(C))
*210                      *QD(C)**(-rhoq(C)))**(-1/rhoq(C));
211
*212 ARMMN(C)$CNM(C)..  QQ(C) =E= QD(C);
213
*214 CET(C)$CE(C)..     QX(C) =E= gammat(C)*
*215                      (deltat(C)*QE(C)**rhot(C) + (1-deltat(C))
*216                      *QD(C)**rhot(C))**(1/rhot(C));
217
*218 CETNE(C)$CNE(C)..  QX(C) =E= QD(C);
219
*220 EDRATIO(C)$CE(C)..  QE(C)/QD(C) =E=
*221                      (PE(C)/PD(C)*(1-deltat(C)) / deltat(C))
*222                      **(1/(rhot(C)-1));
223
*224 IDRATIO(C)$CM(C)..  QM(C)/QD(C) =E=
*225                      ((PD(C)/PM(C))*(deltaq(C)/(1-deltaq(C))))
*226                      **(1/(1+rhoq(C)));
227
228
229 *INSTITUTION BLOCK+++++++
230
231 FACTRNS(H,F)..      YF(H,F) =E=
232                      shry(H,F)*SUM(A, WF(F)*WFDIST(F,A)*QF(F,A));
233
234 GOVEXP..             EG =E= SUM(C, PQ(C)*qg(C)) + SUM(H, tr(H,'GOV'));
235
236 GOVREV..             YG =E= SUM(H, ty(H)*YH(H))
*237                     + EXR*tr('GOV', 'ROW')
*238                     + SUM(C, tq(C)*(PD(C)*QD(C)+(PM(C)*QM(C))$CM(C)))

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

*239          + SUM(C$CM(C), tm(C)*EXR*pwm(C)*QM(C))
*240          + SUM(C$CE(C), te(C)*EXR*pwe(C)*QE(C));
241
242  HHINC(H)..  YH(H) =E= SUM(F, YF(H,F)) + tr(H,'GOV')
*243          + EXR*tr(H,'ROW');
244
245  HHDEM(C,H).. QH(C,H) =E=
246          alpha(C,H)*(1 - MPS(H))*(1 - ty(H))*YH(H)/PQ(C);
247
248  INVDEM(C)..  QINV(C) =E= qinvbar(C)*IADJ;
249
250
251  *SYSTEM CONSTRAINT BLOCK+++++++
252
253  FACTEQ(F)..  SUM(A, QF(F,A)) =E= QFS(F);
254
255  COMEQ(C)..  QQ(C) =E= SUM(A, QINT(C,A)) + SUM(H, QH(C,H))
256          + qg(C) + QINV(C);
257
*258  CURACC..  SUM(C$CE(C), pwe(C)*QE(C)) + SUM(I, tr(I,'ROW'))
*259          + FSAV =E= SUM(C$CM(C), pwm(C)*QM(C));
260
261  SAVINV..  SUM(H, MPS(H)*(1 - ty(H))*YH(H)) + (YG - EG)
*262          + EXR*FSAV =E= SUM(C, PQ(C)*QINV(C)) + WALRAS;
263
264  PNORM..  SUM(C, PQ(C)*cwts(C)) =E= cpi;
265
266
267  *MODEL=====
268
269  MODELS
270
271  AGE5  Open-economy model
272  /ALL/
273  ;
274
275  *SOCIAL ACCOUNTING MATRIX=====
276
277  TABLE SAM(AC,ACP)  social accounting matrix
278
279          FA  FB  CA  CB  K  L  HA  HB  GOV
280  FA          279
281  FB          394
282  CA      84  55          30  49  13
283  CB      50  99          165  92  67
284  K       73  135
285  L       72  105
286  HA          125  95          25

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

287 HB                      83  82                      5
288 GOV
289 S-I                      70   40  -1
290 YTAX                      20   5
291 STAX                      10   20
*292 TAR                      39
*293 ROW                      105
294
*295 +      S-I  YTAX  STAX  TAR  ROW
296 FA
297 FB
298 CA      28                      30
299 CB      85
300 K
301 L
302 HA                      40
303 HB                      16
304 GOV      25   30   39  15
305 S-I                      4
306 YTAX
307 STAX
*308 TAR
*309 ROW
310 ;
311
312 PARAMETER
313   tdiff(AC) column minus row total for account ac;
314 *This parameter is used to check that the above SAM is balanced.
315   SAM('TOTAL',ACNTP) = SUM(ACNT, SAM(ACNT,ACNTP));
316   SAM(ACNT,'TOTAL') = SUM(ACNTP, SAM(ACNT,ACNTP));
317   tdiff(ACNT)       = SAM('TOTAL',ACNT) - SAM(ACNT,'TOTAL');
318
319 DISPLAY SAM, tdiff;
320
321
322 *ASSIGNMENTS FOR PARAMETERS AND VARIABLES=====
323
324 PARAMETERS
325 *The following parameters are used to define initial values of
326 *model variables.
*327 EGO, IADJO, MPSO(H), EXRO, FSAVO,
*328 PAO(A), PVAO(A), PXO(C),
*329 PDO(C), PEO(C), PMO(C), PQO(C),
330 QAO(A), QFO(F,A), QFSO(F), QHO(C,H), QINTO(C,A), QINV0(C),
*331 QDO(C), QEO(C), QMO(C), QQO(C), QXO(C),
332 WFO(F), WFDISTO(F,A), YFO(H,F), YGO, YHO(H)
333 ;
334

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

335
336 *FACTOR EMPLOYMENT AND PRICES+++++++
337
338 PARAMETERS
339 labor(A)      no. of workers employed by firm a
340               /FA 100, FB 50/
341
342 wfa(F,A)      calibrated factor f's cost in firm a
343 cdiff(F,A)    difference btwn wfa and SAM value
344 ;
345
346 *Defining factor employment and supply
347 QF0('K',A) = SAM('K',A);
348 QF0('L',A) = labor(A);
349 QFS0(F)     = SUM(A, QF0(F,A));
350
351 *Computing firm-specific wage
352 wfa(F,A)    = SAM(F,A)/QF0(F,A);
353
354 *Computing average wage
355 WF0(F)      = SUM(A, SAM(F,A))/SUM(A, QF0(F,A));
356
357 *Computing wage distortion factors
358 WFDIST0(F,A) = wfa(F,A) / WF0(F);
359
360 *Checking calibration
361 cdiff(F,A) = WF0(F)*WFDIST0(F,A)*QF0(F,A) - SAM(F,A);
362
363 DISPLAY wfa, cdiff;
364
365
366 *PRICE BLOCK+++++++
367
368 PARAMETERS
369 sigmaq(C)   elas. of subst. bt. dom goods and imports for c
370 sigmat(C)   elas. of trans. bt. dom sales and exports for c
371 ;
372
373 EXRO        = 1;
374 PA0(A)      = 1;
375 PD0(C)      = 1;
376 PE0(C)      = 1;
377 PH0(C)      = 1;
378 PX0(C)      = 1;
379 PVA0(A)     = SUM(F, SAM(F,A)) / (SAM(A, 'TOTAL')/PA0(A));
380 tq(C)       = SAM('STAX',C)
381             /((SAM('TAR',C) + SAM('ROW',C) + SUM(A, SAM(A,C))
382             - SAM(C, 'ROW') );

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

383 PQ0(C)      = 1 + tq(C);
384 QA0(A)      = SAM('TOTAL',A)/PA0(A);
*385 QD0(C)     = (SUM(A, SAM(A,C)) - SAM(C,'ROW'))/PD0(C);
*386 QE0(C)     = SAM(C,'ROW')/PE0(C);
*387 QM0(C)     = (SAM('ROW',C) + SAM('TAR',C))/PM0(C);
*388 QO0(C)     = (SAM('TOTAL',C) - SAM(C,'ROW'))/PQ0(C);
*389 QX0(C)     = SUM(A, SAM(A,C))/PX0(C);
390 ica(C,A)    = (SAM(C,A)/PQ0(C)) / QA0(A);
391 theta(A,C)  = (SAM(A,C)/PX0(C)) / QA0(A);
*392 te(C)      = 0;
*393 pwe(C)      = PE0(C)/((1 + te(C))*EXR0);
*394 tm(C)$CM(C) = SAM('TAR',C)/SAM('ROW',C);
*395 pwm(C)$CM(C) = PM0(C) / ( EXR0*(1 + tm(C)) );
396
397
398 *PRODUCTION AND COMMODITY BLOCK+++++++
399
400 QINT0(C,A)   = SAM(C,A)/PQ0(C);
401 beta(F,A)    = SAM(F,A) / SUM(FP, SAM(FP,A));
402 lambda(A)    = QA0(A) / PROD(F, QF0(F,A)**beta(F,A));
*403 sigmat(C)   = 2.0;
*404 sigmaq(C)   = 0.7;
*405 rhot(C)     = 1/sigmat(C) + 1;
*406 rhoq(C)     = 1/sigmaq(C) - 1;
*407 deltat(C)$CE(C) = 1/(1 + (PD0(C)/PE0(C))*
*408             (QE0(C)/QD0(C)**(rhot(C)-1));
*409 deltaq(C)$CM(C) = 1/(1 + (PD0(C)/PM0(C))*
*410             (QD0(C)/QM0(C)**(1+rhoq(C)));
*411 gammat(C)$CE(C) = QX0(C) / ( deltat(C)*QE0(C)**rhot(C)
*412             + (1-deltat(C))*QD0(C)**rhot(C) )**(1/rhot(C));
*413 gammaq(C)$CM(C) = QQ0(C) / ( deltaq(C)*QM0(C)**(-rhoq(C))
*414             + (1-deltaq(C))*QD0(C)**(-rhoq(C)) )**(-1/rhoq(C));
415
416
417 *INSTITUTION BLOCK+++++++
418
419 EGO          = SAM('TOTAL', 'GOV') - SAM('S-I', 'GOV');
*420 FSAVO      = SAM('S-I', 'ROW')/EXR0;
421 IADJO       = 1;
422 MPS0(H)     = SAM('S-I',H) / (SAM('TOTAL',H) - SAM('YTAX',H));
423 QH0(C,H)    = SAM(C,H)/PQ0(C);
424 QINVO(C)    = SAM(C,'S-I')/PQ0(C);
425 YF0(H,F)    = SAM(H,F);
426 YGO        = SAM('GOV', 'TOTAL');
427 YH0(H)      = SAM('TOTAL',H);
428 alpha(C,H)  = SAM(C,H)/SUM(CP, SAM(CP,H));
429 qg(C)       = SAM(C,'GOV')/PQ0(C);
430 qinvbar(C)  = SAM(C,'S-I')/PQ0(C);

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

431 shry(H,F) = SAM(H,F) / SAM('TOTAL',F);
432 tr(H,'GOV') = SAM(H,'GOV');
*433 tr(I,'ROW') = SAM(I,'ROW')/EXR0;
434 ty(H) = SAM('YTAX',H) / SAM('TOTAL',H);
435
436
437 *SYSTEM CONSTRAINT BLOCK+++++++
438
439 cwts(C) = SUM(H, SAM(C,H)) / SUM((CP,H), SAM(CP,H));
440 cpi = SUM(C, cwts(C)*PQ0(C));
441
442
443 *INITIALISATION=====
444
445 EG.L = EGO;
*446 EXR.L = EXR0;
*447 FSAV.L = FSAV0;
448 IADJ.L = IADJ0;
449 MPS.L(H) = MPS0(H);
450 PA.L(A) = PA0(A);
*451 PD.L(C) = PD0(C);
*452 PE.L(C) = PE0(C);
*453 PM.L(C) = PM0(C);
*454 PQ.L(C) = PQ0(C);
455 PVA.L(A) = PVA0(A);
456 PX.L(C) = PX0(C);
457 QA.L(A) = QA0(A);
*458 QD.L(C) = QD0(C);
*459 QE.L(C) = QE0(C);
460 QF.L(F,A) = QF0(F,A);
461 QFS.L(F) = QFS0(F);
462 QH.L(C,H) = QH0(C,H);
463 QINT.L(C,A) = QINT0(C,A);
464 QINV.L(C) = QINV0(C);
*465 QM.L(C) = QM0(C);
*466 QQ.L(C) = QQ0(C);
*467 QX.L(C) = QX0(C);
468 WF.L(F) = WF0(F);
469 WFDIST.L(F,A) = WFDIST0(F,A);
470 YF.L(H,F) = YF0(H,F);
471 YG.L = YG0;
472 YH.L(H) = YH0(H);
473
474 *DISPLAY+++++++
475
476 DISPLAY
*477 cpi, cwts, ica, qg, qinvbar, pwe, pwm,
*478 shry, tr, ty, tq, te, tm,

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

*479 alpha, beta, gammaq, gammat, deltaq, deltat,
*480 lambda, rhoq, rhot, sigmaq, sigmat, theta,
*481 EG.L, EXR.L, FSAV.L, IADJ.L, MPS.L,
*482 PA.L, PD.L, PE.L, PM.L, PQ.L, PVA.L, PX.L,
*483 QA.L, QD.L, QE.L, QF.L, QFS.L, QH.L, QINT.L, QINV.L,
*484 QM.L, QQ.L, QX.L, WF.L, WFDIST.L, YF.L, YG.L, YH.L
485 ;
486
487
488 *CLOSURE RULES+++++
489
490 *SAVINGS-INVESTMENT BALANCE
491 *investment-driven savings
492 IADJ.FX      = IADJO;
493 MPS.LO('HA') = -INF;
494 MPS.UP('HA') = +INF;
495 MPS.L('HA')  = MPSO('HA');
496 MPS.FX('HB') = MPSO('HB');
497
498 *savings-driven investment
499 * MPS.FX(H)  = MPSO(H);
500 * IADJ.LO    = -INF;
501 * IADJ.UP    = +INF;
502 * IADJ.L     = IADJO;
503
504
505 *CAPITAL MARKET
506 *capital is firm-specific and fully employed
507 WFDIST.LO('K',A) = -INF;
508 WFDIST.UP('K',A) = +INF;
509 WFDIST.L('K',A)  = WFDISTO('K',A);
510 WF.FX('K')       = WFO('K');
511 QF.FX('K',A)     = QFO('K',A);
512 QFS.LO('K')      = -INF;
513 QFS.UP('K')      = +INF;
514 QFS.L('K')      = QFSO('K');
515
516 *capital is mobile and fully employed
517 * WFDIST.FX('K',A) = WFDISTO('K',A);
518 * WF.LO('K')      = -INF;
519 * WF.UP('K')      = +INF;
520 * WF.L('K')      = WFO('K');
521 * QF.LO('K',A)    = -INF;
522 * QF.UP('K',A)    = +INF;
523 * QF.L('K',A)    = QFO('K',A);
524 * QFS.FX('K')    = QFSO('K');
525
526

```


GAMS による応用一般均衡分析：海外部門のあるモデル

```

527 *LABOR MARKET
528 *labor is mobile and unemployed (fixed wages)
529 WFDIST.FX('L',A) = WFDIST0('L',A);
530 WF.FX('L')      = WFO('L');
531 QF.LO('L',A)    = -INF;
532 QF.UP('L',A)    = +INF;
533 QF.L('L',A)     = QFO('L',A);
534 QFS.LO('L')     = -INF;
535 QFS.UP('L')     = +INF;
536 QFS.L('L')     = QFS0('L');
537
538 *labor is mobile and fully employed
539 * WFDIST.FX('L',A) = WFDIST0('L',A);
540 * WF.LO('L')      = -INF;
541 * WF.UP('L')      = +INF;
542 * WF.L('L')      = WFO('L');
543 * QF.LO('L',A)    = -INF;
544 * QF.UP('L',A)    = +INF;
545 * QF.L('L',A)    = QFO('L',A);
546 * QFS.FX('L')    = QFS0('L');
547
548
*549 *FOREIGN EXCHANGE MARKET
*550 *foreign savings is fixed.
*551 *exchange rate is flexible and clears B/P.
*552 FSAV.FX = FSAV0;
*553 EXR.LO = -INF;
*554 EXR.UP = +INF;
*555 EXR.L = EXR0;
556
*557 *exchange rate is fixed.
*558 *foreign savings is flexible and clears B/P.
*559 * EXR.FX = EXR0;
*560 * FSAV.LO = -INF;
*561 * FSAV.UP = +INF;
*562 * FSAV.L = FSAV0;
563
564
565 *SOLVE STATEMENT FOR BASE=====
566
567 AGES.HOLDFIXED = 1;
568
569 *SOLVE AGES USING MCP;
570
571
572 *REPORT SETUP AND BASE REPORT=====
573
574 *SET AND PARAMETERS FOR REPORTS+++++

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

575
576 SET
577 SIM          simulations
578 /BASE      base simulation
*579 HYPO      25% increase in CA export price/
580
*581 ACGDP          GDP items
*582 /GDPMP1      GDP at market prices (from spending side)
*583 PRVCON        private consumption
*584 GOVCON        government consumption
*585 INVEST        investment
*586 EXP           exports of goods and services
*587 IMP           imports of goods and services
*588 NITAX         net indirect taxes
*589 GDPFC        GDP at factor prices
*590 GDPMP2      GDP at market prices (from income side)
*591 GDGPAP      gap bt GDPMP1 and GDPMP2/
592
*593 ACGDP1(ACGDP) components of GDP at market prices
*594 /PRVCON      private consumption
*595 GOVCON        government consumption
*596 INVEST        investment
*597 EXP           exports of goods and services
*598 IMP           imports of goods and services/
599 ;
600
601 PARAMETERS
602
*603 PWEASIM(SIM)   CA export price (for curr) (experiment par.)
*604 PWEREP(C,SIM)  export price for commodity c (value used)
605 EGREP(SIM)       government expenditures
*606 EXRREP(SIM)    exchange rate (dom. cur. per unit of for. cur.)
*607 FSAVREP(SIM)   foreign savings (foreign currency)
*608 GDPREP(*,SIM)  nominal GDP data
609 IADJREP(SIM)     investment adjustment factor
610 MPSREP(H,SIM)   mps (=aps) for household h
611 PAREP(A,SIM)     price of firm a
*612 PDREP(C,SIM)   domestic price of domestic output c
*613 PEREP(C,SIM)   export price for c (domestic currency)
*614 PMREP(C,SIM)   import price for c (domestic currency)
*615 PQREP(C,SIM)   composite commodity price for c
616 PVAREP(A,SIM)   value-added price for firm a
617 PXREP(C,SIM)     producer price for commodity c
618 QAREP(A,SIM)     production level of firm a
*619 QDREP(C,SIM)   quantity sold domestically of domestic output c
*620 QEREP(C,SIM)   quantity of exports for commodity c
621 QFREP(F,A,SIM)   firm a's demand of factor f
622 QFSREP(F,SIM)   supply of factor f

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

623 QHREP(C,H,SIM)    household h's demand of commodity c
624 QINTREP(C,A,SIM)  firm a's intermediate input of commodity c
625 QINVREP(C,SIM)    quantity of investment by commodity of origin c
*626 QMREP(C,SIM)     quantity of imports for commodity c
*627 QQREP(C,SIM)    quantity of domestic composite supply for c
*628 QXREP(C,SIM)    quantity of domestic output for c
629 WALRASREP(SIM)    dummy variable (zero at equilibrium)
630 WFREP(F,SIM)      average price of factor f
631 WFAREP(F,A,SIM)   factor f's price for firm a
632 WFDISTREP(F,A,SIM) factor f's price distortion in firm a
633 YFREP(H,F,SIM)    household h's income from factor f
634 YGREP(SIM)        government revenue
635 YHREP(H,SIM)      income of household h
636 SAMREP(SIM,AC,ACP) SAM computed from model solution
637 SAMCHK(AC,SIM)    column minus row total for account ac in SAM
638 ;
639
*640 PWEASIM('BASE') = pwe('CA');
641
*642 *25% increase in CA export price
*643 PWEASIM('HYPO') = 1.25*pwe('CA');
644
*645 DISPLAY PWEASIM;
646
647
648 LOOP(SIM,
649
*650 pwe('CA') = PWEASIM(SIM);
651
652 SOLVE AGE5 USING MCP;
653
*654 PWEREP(CE,SIM)    = pwe(CE);
655 EGREP(SIM)         = EG.L;
*656 EXRREP(SIM)      = EXR.L;
*657 FSAVREP(SIM)    = FSAV.L;
658 IADJREP(SIM)     = IADJ.L;
659 MPSREP(H,SIM)    = MPS.L(H);
660 PAREP(A,SIM)     = PA.L(A);
*661 PDREP(C,SIM)    = PD.L(C);
*662 PEREP(CE,SIM)   = PE.L(CE);
*663 PMREP(CM,SIM)   = PM.L(CM);
*664 PQREP(C,SIM)    = PQ.L(C);
665 PVAREP(A,SIM)    = PVA.L(A);
*666 PXREP(C,SIM)    = PX.L(C);
667 QAREP(A,SIM)     = QA.L(A);
*668 QDREP(C,SIM)    = QD.L(C);
*669 QEREP(CE,SIM)   = QE.L(CE);
*670 QFREP(F,A,SIM) = QF.L(F,A);

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

*671 QFSREP(F,SIM)      = QFS.L(F);
*672 QHREP(C,H,SIM)    = QH.L(C,H);
*673 QINTREP(C,A,SIM)  = QINT.L(C,A);
*674 QINVREP(C,SIM)    = QINV.L(C);
*675 QMREP(CM,SIM)     = QM.L(CM);
*676 QQREP(C,SIM)      = QQ.L(C);
*677 QXREP(C,SIM)      = QX.L(C);
*678 WALRASREP(SIM)    = WALRAS.L;
*679 WFREP(F,SIM)      = WF.L(F);
*680 WFAREP(F,A,SIM)   = WF.L(F)*WFDIST.L(F,A);
*681 WFDISTREP(F,A,SIM) = WFDIST.L(F,A);
*682 YFREP(H,F,SIM)    = YF.L(H,F);
*683 YGREP(SIM)         = YG.L;
*684 YHREP(H,SIM)      = YH.L(H);
*685
*686
*687 *GDP data
*688 GDPREP('PRVCON',SIM) = SUM((C,H), PQ.L(C)*QH.L(C,H)) ;
*689 GDPREP('GOVCON',SIM) = SUM(C, PQ.L(C)*qg(C));
*690 GDPREP('INVEST',SIM) = SUM(C, PQ.L(C)*QINV.L(C));
*691 GDPREP('EXP',SIM)     = SUM(C, EXR.L*pwe(C)*QE.L(C));
*692 GDPREP('IMP',SIM)    = - SUM(C, EXR.L*pwm(C)*QM.L(C));
*693 GDPREP('GDPFC',SIM)  = SUM((F,A), WF.L(F)*WFDIST.L(F,A)*QF.L(F,A));
*694 GDPREP('NITAX',SIM)
*695 = SUM(C, tq(C)*(PD.L(C)*QD.L(C) + (PM.L(C)*QM.L(C))*$CM(C)))
*696 + SUM(C,$CM(C), tm(C)*EXR.L*pwm(C)*QM.L(C))
*697 + SUM(C,$CE(C), te(C)*EXR.L*pwe(C)*QE.L(C));
*698
*699 *SAM items
*700 *Factor payments
*701 SAMREP(SIM,F,A) = WFDIST.L(F,A)*WF.L(F)*QF.L(F,A);
*702 *Payments from commodities
*703 SAMREP(SIM,A,C) = PX.L(C)*theta(A,C)*QA.L(A);
*704 *Payments from factors
*705 SAMREP(SIM,H,F) = YF.L(H,F);
*706 *Household demands
*707 SAMREP(SIM,C,H) = PQ.L(C)*QH.L(C,H);
*708 *Government demands
*709 SAMREP(SIM,C,'GOV') = PQ.L(C)*qg(C);
*710 *Intermediate demands
*711 SAMREP(SIM,C,A) = PQ.L(C)*ica(C,A)*QA.L(A);
*712 *Savings supplies
*713 SAMREP(SIM,'S-I',H) = MPS.L(H)*(1-ty(H))*YH.L(H);
*714 *Investment demands
*715 SAMREP(SIM,C,'S-I') = PQ.L(C)*IADJ.L*qinvbar(C);
*716 *Export
*717 SAMREP(SIM,CE,'ROW') = PE.L(CE)*QE.L(CE);
*718 *Import

```

GAMS による応用一般均衡分析：海外部門のあるモデル

```

*719 SAMREP(SIM,'ROW',CM) = EXR.L*pwm(CM)*QM.L(CM);
*720 *Tariff
*721 SAMREP(SIM,'TAR',CM) = tm(CM)*EXR.L*pwm(CM)*QM.L(CM);
722 *Sales tax from commodity
723 SAMREP(SIM,'STAX',C) = tq(C)*(SUM(A,SAMREP(SIM,A,C))
724 -SAMREP(SIM,C,'ROW')+SAMREP(SIM,'ROW',C)+SAMREP(SIM,'TAR',C));
725 *Transfer from gov to households
726 SAMREP(SIM,H,'GOV') = tr(H,'GOV');
727 *Income tax from households
728 SAMREP(SIM,'YTAX',H) =
729 ty(H)*(SUM(F,YF.L(H,F))+tr(H,'GOV')+EXR.L*tr(H,'ROW'));
730 *Income tax revenue
731 SAMREP(SIM,'GOV','YTAX') = SUM(H,SAMREP(SIM,'YTAX',H));
732 *Sales tax revenue
733 SAMREP(SIM,'GOV','STAX') = SUM(C,SAMREP(SIM,'STAX',C));
*734 *Tariff revenue
*735 SAMREP(SIM,'GOV','TAR') = SUM(CM,SAMREP(SIM,'TAR',CM));
*736 *Transfers from ROW
*737 SAMREP(SIM,H,'ROW') = EXR.L*tr(H,'ROW');
*738 SAMREP(SIM,'GOV','ROW') = EXR.L*tr('GOV','ROW');
739 *Government savings
740 SAMREP(SIM,'S-I','GOV') =
741 (SAMREP(SIM,'GOV','YTAX')+SAMREP(SIM,'GOV','STAX')
742 +SAMREP(SIM,'GOV','TAR')+SAMREP(SIM,'GOV','ROW'))
743 -(SUM(C,SAMREP(SIM,C,'GOV))+SUM(H,SAMREP(SIM,H,'GOV')));
*744 *Foreign savings
*745 SAMREP(SIM,'S-I','ROW') = EXR.L*FSAV.L;
746 );
747
748 *Computing totals for SAMREP
749 SAMREP(SIM,'TOTAL',ACNTP) = SUM(ACNT, SAMREP(SIM,ACNT,ACNTP));
750 SAMREP(SIM,ACNT,'TOTAL') = SUM(ACNTP, SAMREP(SIM,ACNT,ACNTP));
751 *Check that SAMREP is balanced
752 SAMCHK(ACNT,SIM) = SAMREP(SIM,'TOTAL',ACNT) - SAMREP(SIM,ACNT,'TOTAL');
753
*754 *Processing GDP data
*755 GDPREP('GDPMP1',SIM) = SUM(ACGDP1, GDPREP(ACGDP1,SIM));
*756 GDPREP('GDPMP2',SIM) = GDPREP('GDPFC',SIM) + GDPREP('NITAX',SIM);
*757 GDPREP('GDGPAP',SIM) = GDPREP('GDPMP1',SIM) - GDPREP('GDPMP2',SIM);
758
759 OPTION QFREP:3:1:1, QHREP:3:1:1, QINTREP:3:1:1, WFAREP:3:1:1,
760 WFAREP:3:1:1, WFDISTREP:3:1:1, YFREP:3:1:1, SAMREP:3:1:1;
761
762 DISPLAY
*763 PWREP, EGREP, EXRREP, FSAVREP, IADJREP, MPSREP, PAREP, PDREP, PEREP,
*764 PMREP, PQREP, PVAREP, PXREP, QAREP, QDREP, QEREP, QFREP, QFSREP, QHREP,
*765 QINTREP, QINVREP, QMREP, QQREP, QXREP, WFREP, WFAREP, WFDISTREP, YFREP,
*766 YGREP, YHREP, WALRASREP, GDPREP, SAMREP, SAMCHK
767 ;

```

海外部門を取り入れて貿易を分析対象としたモデル拡張により、age 5 モデルの GAMS コードは age 4 モデルからどのように変更されるか、入力ファイルの変更箇所を見ていこう。

社会勘定行列 SAM には、海外部門 ROW と関税 TAR という 2 つの勘定項目が新たに加えられる。そのために、集合の定義 (18 行以下) の中で勘定項目を表す大域集合 AC の要素にこれら 2 つを追加する (35-36 行)。また、経済主体 I の集合にも、海外部門 ROW という成分が追加される (63 行)。以上の準備の後に、社会勘定行列を定義する (275-310 行)。

さらに、国内で生産され国内で販売される国産財、国内で生産され海外に販売される輸出財、そして海外で生産され国内で販売される輸入財の間の不完全代替性や不完全変形可能性を考慮するために、商品を輸入財 CE、非輸入財 NCE、輸出財 CM、非輸出財 NCM のグループに分ける必要がある (47-57 行)。これらのグループは商品の集合 C の部分集合になる。ただし、現在の age 5 モデルには 2 種類しか商品がないために、各グループに属する商品の数はそれぞれ 1 種類である。

次に、新たなパラメーターを見てみよう。貿易を取り入れるというモデル拡張により、パラメーターとしては輸出財の国際価格 (外貨建て) pwe (C)、輸入財の国際価格 (外貨建て) pwm (C) (80-81 行)、輸出関税率 te (C)、輸入関税率 tm (C) (85-86 行)、Armington 関数の規模係数 $gammaq$ (C)、加重 $deltaq$ (C)、指数 $rhoq$ (C)、輸出変形関数の規模係数 $gammat$ (C)、加重 $deltat$ (C)、指数 $rhott$ (C) (93-96 行、98-99 行) が追加される。また変数としては新たに、為替レート (外貨 1 単位当たりの自国通貨) EXR と海外部門貯蓄 (外貨建て) FSAV (108-109 行)、自国が生産する商品 c の国内価格 PD (C)、輸出価格 (自国通貨建て) PE (C)、輸入価格 (自国通貨建て) PM (C)、合成財 c の国内供給価格 PQ (C) (113-116 行)、自国が生産する商品 c の国内供給量 QD (C)、輸出量 QE (C) (120-121 行)、商品 c の輸入量 QM (C)、国内生産量 QQ (C)、合

成財 c の国内供給量 $QX(C)$ (127-129 行) が追加される。

さらに、変数の初期値を表すパラメーターが 327, 329, 331 行で、また国産財と輸入財の代替の弾力性 $\sigma_{aq}(C)$ と国内供給と輸出の変形の弾力性 $\sigma_{mat}(C)$ が 369-370 行で宣言され、373-389 行で初期値が代入される。これらの中で特に注目に値するのは、輸出関税率 $te(C)=0$ 、国産財と輸入財の代替の弾力性 $\sigma_{aq}(C)=2.0$ 、国内供給と輸出の代替の弾力性 $\sigma_{mat}(C)=0.7$ 、投資方程式の比例的調整係数 $IADJ0=1$ である。これらの初期値および基準均衡の社会勘定行列に基づいて、その他のパラメーターの値および変数の初期値がカラブレイトされる (379-402 行, 405-420 行, 422-434 行)。

この age 5 モデルを構成する方程式体系には、前稿の age 4 モデルと比べて、いくつかの式が新たに追加されている (第 3 節参照)。入力ファイルでは、新たに導入された方程式については、方程式の名称宣言と定義という手続きが必要とされる。新たな方程式の導入による入力ファイルの変更箇所を表 4 にまとめた。

モデルの閉じ方では、(1) 貯蓄—投資バランスについて、投資主導型で貯蓄が調整される (491-496 行) か、あるいは貯蓄主導型で投資が調整される (498-502 行) か、(2) 要素市場に関して、資本は企業間を移動可能である (516-524 行) か、あるいは資本市場は企業別に成立しており企業間移動は不可能である (506-514 行) か、(3) 労働については、賃金率が固定されており失業が発生している (528-536 行) か、あるいは伸縮的な賃金率の下で完全雇用状態が成立している (538-546 行) か、また (4) 外国為替市場については、為替レートは伸縮的で海外部門貯蓄が固定されている (550-555 行) か、あるいはその逆で為替レートは固定的で海外部門貯蓄が伸縮的である (557-562 行) かを区別する。ただし、このうちの (1)-(3) は前稿の age 4 モデルでも検討しており、外国為替市場についての (4) の区別が新しい。モデルの閉じ方の組み合わせにより、モデル類型は表 5 にま

表 4 age 5 モデルにおける新たな方程式

モデル 方程式	入力ファイル		
	方程式名	宣言	定義
(1) 輸入価格	PMDEF	143 行	180 行
(2) 輸出価格	PEDEF	144 行	182 行
(3) 国内総支出額	ABSORD	145 行	184 行
(4) 輸出可能財の国内総生産額	OUTVAL	146 行	188-189 行
(11) Armington 関数 (輸入財)	ARM	155 行	208-210 行
(12) 輸入品一国産品需要比率	IDRATIO	144 行	224-226 行
(13) Armington 関数 (非輸入財)	ARMNM	156 行	212 行
(14) 合成供給関数 (輸入財)	CET	157 行	214-216 行
(15) 輸出一国内供給比率	EDRATIO	159 行	220-222 行
(16) 合成供給関数 (非輸入財)	CETNE	158 行	218 行
(24) 商品市場のバランス条件	COMAQ	172 行	255-256 行
(25) 経常収支のバランス条件	CURACC	173 行	258-259 行
(26) 資本市場のバランス条件	SAVINV	174 行	261-262 行

とめたように 16 通りになる。表 3 はモデル類型 (a) の入力ファイルであるが、該当箇所の行頭の* (アスタリスク) を移動することによって他のモデル類型の入力ファイルに変更することができる。

565 行以下が解法である。ここでは、LOOP 命令を利用して、基準均衡 BASE と仮設均衡 HYPO (商品 CA の国際価格 pwe (CA) が 25% 上昇する場合) の 2 通りの均衡解を求めている。そのために、577-579 行の SET 命令で LOOP を回すための添え字 SIM を、また 601-638 行の PARAMETER 命令で変数を準備する。また、age 5 モデルでは新たに国内総生産 GDP も求めることにして、その準備を 581-599 行で行っている。LOOP 文は 648 行から 746 行にわたっており、途中の 652 行に SOLVE 命令がある。LOOP 文の中の、687-697 行は国内総生産 GDP を、699-745 行は社会勘定行列を求めるための計算である。LOOP 文に続く 748-752 行は求めた社会勘定行列の整合性の確認、754-757 行は国内総生産 GDP の計算である。762-767 行の DISPLAY 命令は、基準均衡と仮設均衡の均衡

GAMS による応用一般均衡分析：海外部門のあるモデル

表5 モデルの閉じ方と類型 (age 5)

モデル 類型	貯蓄—投資バランス		要素市場				外国為替市場	
	投資主導型 貯蓄	貯蓄主導型 投資	資本		労働		為替レート	
			企業別 資本市場	企業間 移動自由	企業間移動自由		伸縮的	固定
			完全雇用 伸縮的貸貸率		失業 固定貸金率	完全雇用 伸縮的貸金率	海外貯蓄	
						固定	伸縮的	
a	○		○		○		○	
b	○			○	○		○	
c	○		○			○	○	
d	○			○		○	○	
e		○	○		○		○	
f		○		○	○		○	
g		○	○			○	○	
h		○		○		○	○	
i	○		○		○			○
j	○			○	○			○
k	○		○			○		○
l	○			○		○		○
m		○	○		○			○
n		○		○	○			○
o		○	○			○		○
p		○		○		○		○

入力ファイル

該当箇所 491-496行 498-502行 506-514行 516-524行 528-536行 538-546行 550-555行 557-562行

値の比較を出力する。

6. 結果のまとめ

前節の入力ファイル (表3) を数理計画法解法プログラム GAMS でコンパイルし実行すると、出力ファイルが得られる。出力ファイルは長いのでここには掲載しないが、注意すべき箇所について出力ファイルの見方を説明しよう。

各モデルタイプの出力ファイルの最初の部分は、入力ファイルのオウム返し出力 (エコープリント) である。エコープリントが終わると、GAMS の処理結果が記されている。最初に注目する箇所は、基準均衡の入力チェ

ック（入力ファイル 312-319 行）の処理結果である。社会勘定行列の形でなされた基準均衡のデータ入力（275-310 行）に対して、列和と行和の差 `tdiff` を計算し（317 行）、これらが全て 0 になるかどうかで、正しく入力されたかどうかを判定しようとしている。入力ファイル 319 行の「`tdiff` を示せ」という命令に対する出力ファイルを見ると、結果は「全て 0」となっており正しく入力されていることが分かる。

次に注目する箇所は、入力ファイル 476-485 行の `DISPLAY` 命令によって出力される要素価格、要素価格の歪み係数、商品価格以外のパラメーターのカラブレイションの結果である（表 6 参照）。ここには変数の初期値（基準均衡値）もあわせて出力されている。基準均衡の値は初期値として与えられているので、わざわざ計算の手間を掛ける必要はないと思われるかも知れないが、これらの値は後に示される `LOOP` 文の 1 巡目 `BASE` の計算結果（入力ファイル 652 行の `SOLVE` 命令による計算結果）一致することを確認する（再現テスト）ために必要である。

出力ファイルでは、入力ファイル 648-748 行の `LOOP` 文の中の `SOLVE` 命令（652 行）に対する状況報告において、`SOLVER STATUS`（解法プログラムの状態）が `NORMAL COMPLETION`（正常終了）となっており、`MODEL STATUS`（モデルの状態）が `OPTIMAL`（最適解を見つけて終了）となっていることを確認することが重要である。

`LOOP` 文の 1 巡目 `BASE`（基準均衡）は、パラメーターに基準均衡値を与えた場合の均衡値の計算、すなわち再現テストである。表 7 にまとめたシミュレーション結果を見ると、再現テストに合格していることが分かり、モデルの定式化に成功したと判断される。2 巡目の `HYP0`（仮設均衡）は、輸出品の国際価格 `pwe`（CA）が 25% 上昇することを想定した場合の均衡値を計算している。入力ファイルの 699-745 行で求め、748-752 行でその整合性を確認した仮設均衡の社会勘定行列は表 8 にまとめた⁷⁾。また、687-697 行と 754-757 行で求めた国内総生産 GDP の値は表 7 の最下段に

GAMS による応用一般均衡分析：海外部門のあるモデル

表6 カラブライツされたパラメーター値

tdiff	社会勘定行列の行和と列和の差 全て	0.000	te (C)	商品 c の輸出補助率 CA CB	0.000 0.000
wfa (F,A)	企業 a の要素 f の要素価格 (K,FA) (K,FB) (L,FA) (L,FB)	1.000 1.000 0.720 2.100	tm (C)	商品 c の輸入関税率 CB	0.371
cdiff	wfa と SAM 値の相違 全て	0.000	alpha (H,C)	家計 h の効用関数における商品 c の係数 (商品 c への支出割合) (HA,CA) (HA,CB) (HB,CA) (HB,CB)	0.154 0.846 0.348 0.652
cpi	消費者価格指数 CPI 1.038		beta (F,A)	企業 a の生産関数における要素 f の係数 (企業 a の付加価値に占める要素 f の割合) (K,FA) (L,FA) (K,FB) (L,FB)	0.503 0.497 0.562 0.437
cwts (C)	消費者価格指数 CPI における商品 c の加重 CA CB	0.235 0.765	gamma (C)	Armington 関数の規模係数 CB	1.718
ica (C,A)	企業 a の産出 1 単位当たりの商品 c の中間投入量 (CA,FA) (CA,FB) (CB,FA) (CB,FB)	0.289 0.134 0.173 0.242	gammat (C)	CET 関数の規模係数 CA	2.565
qg (C)	政府の商品 c の需要 CA CB	12.498 64.599	delta (C)	Armington 関数のシェア係数 CB	0.192
qinvbar (C)	基準年投資需要 CA CB	26.919 81.953	deltat (C)	CET 関数のシェア係数 CA	0.742
pwe	輸出価格 (外貨建て) CA CB	1.000 1.000	lambda (A)	企業 a の生産関数の規模係数 FA FB	3.269 4.507
pwm	輸入価格 (外貨建て) CB	0.729	rho (C)	Armington 関数の指数 CA CB	0.429 0.429
shry (H,F)	要素 f の総要素所得のうち家計 h への分配率 (HA,K) (HB,K) (HA,L) (HB,L)	0.601 0.399 0.537 0.463	rhot (C)	CET 関数の指数 CA CB	1.500 1.500
tr(i,I')	機関 i' から i への移転 (HA,GOV) (HA,ROW) (HB,GOV) (HB,ROW) (GOV,ROW)	25.000 40.000 5.000 16.000 15.000	sigma (C)	国産品と輸入品の代替の弾力性 CA CB	0.700 0.700
ty (H)	家計 h の所得税率 HA HB	0.070 0.027	sigmat (C)	国内販売と輸出の変形の弾力性 CA CB	2.000 2.000
tq (C)	商品 c の売上税率 CA CB	0.040 0.037	theta (A,C)	企業 a の活動水準 1 単位当たりの商品 c の産出量 (FA,CA) (FA,CB) (FB,CA) (FB,CB)	1.000 0.000 0.000 1.000

表7 シミュレーションの結果

	ベンチマーク 均衡	設 置							均 衡		
		a	b	c	d	e	f	g		h	
パラメーター											
輸出品国際価格 pwe (外貨建て)	1.000	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250
商品 CA											
要素費用 wfa											
資本 k	1.000	1.132	1.102	1.080	1.053	1.138	1.103	1.080	1.053	1.053	1.053
企業 FA	1.000	1.080	1.102	1.032	1.053	1.077	1.103	1.032	1.053	1.032	1.032
企業 FB	1.000	0.720	0.720	0.766	0.765	0.720	0.720	0.766	0.765	0.765	0.765
労働 L	2.100	2.100	2.100	2.234	2.232	2.100	2.100	2.234	2.232	2.100	2.232
企業 FB											
変数											
政府支出 EG	110.000	109.752	109.833	109.681	109.761	109.732	109.827	109.680	109.760	109.760	109.760
為替レート EXR	1.000	0.889	0.884	0.875	0.871	0.888	0.882	0.875	0.871	0.871	0.871
海外部門貯蓄 PSAV (外貨建て)	4.000	4.000	4.000	4.000	4.000	4.000	4.000	4.000	4.000	4.000	4.000
投資調整係数 IADI	1.000	1.000	1.000	1.000	1.000	1.086	1.087	1.004	1.007	1.007	1.007
限界貯蓄性向 MPS	0.264	0.230	0.229	0.264	0.261	0.264	0.264	0.264	0.264	0.264	0.264
家計 HA	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221
家計 HB											
企業価格 PA											
企業 FA	1.000	1.041	1.031	1.048	1.038	1.044	1.032	1.048	1.038	1.038	1.038
企業 FB	1.000	1.029	1.036	1.031	1.037	1.028	1.036	1.030	1.037	1.037	1.037
国産品の国内価格 PD											
商品 CA	1.000	1.032	1.022	1.042	1.031	1.035	1.023	1.042	1.031	1.031	1.031
商品 CB	1.000	1.029	1.036	1.031	1.037	1.028	1.036	1.030	1.037	1.037	1.037
輸出価格 PE (自国通貨建て)											
商品 CA	1.000	1.111	1.105	1.094	1.089	1.110	1.103	1.094	1.088	1.088	1.088
商品 CB	1.000										
輸入価格 PM (自国通貨建て)											
商品 CA	1.000	0.889	0.884	0.875	0.871	0.888	0.882	0.875	0.871	0.871	0.871
商品 CB	1.000										
国内供給価格 PQ											
商品 CA	1.040	1.074	1.063	1.084	1.073	1.077	1.064	1.084	1.073	1.073	1.073
商品 CB	1.037	1.027	1.030	1.024	1.027	1.026	1.030	1.024	1.027	1.027	1.027

GAMS による応用一般均衡分析：海外部門のあるモデル

企業付加価値価格 PVA	0.520	0.553	0.546	0.577	0.550	0.555	0.546	0.557	0.550
企業 FA	0.609	0.636	0.636	0.637	0.644	0.635	0.644	0.637	0.644
FB									
生産者価格 PX	1.000	1.041	1.031	1.048	1.038	1.044	1.032	1.048	1.038
商品 CA	1.000	1.029	1.036	1.031	1.037	1.028	1.036	1.030	1.037
CB									
活動水準 QA	279.000	296.705	298.780	281.056	283.546	297.491	299.911	281.075	283.197
企業 FA	394.000	407.522	406.550	388.820	388.200	407.046	405.911	399.771	388.118
FB									
国産品の国内販売量 QD	249.000	260.252	261.737	248.038	249.546	261.163	262.895	248.067	249.613
商品 CA	394.000	407.522	406.550	388.820	388.200	407.046	405.911	388.771	388.118
CB									
輸出量 QE	30.000	36.286	36.850	32.950	33.506	36.176	36.836	32.940	33.499
商品 CA									
要素需要 QF	73.000	73.000	74.495	73.000	74.433	73.000	74.749	73.000	74.454
資本 K	100.000	113.191	112.456	101.490	100.999	113.796	112.925	101.504	101.013
労働 L	135.000	135.000	133.505	135.000	133.567	135.000	133.251	135.000	133.546
資本 K	50.000	54.009	54.489	48.510	49.001	53.865	54.427	48.496	48.987
労働 L									
要素供給 QFS	208.000	208.000	208.000	208.000	208.000	208.000	208.000	208.000	208.000
資本 K	150.000	167.200	166.945	150.000	150.000	167.661	167.352	150.000	150.000
労働 L									
家計需要 OH	28.842	31.034	31.428	28.340	28.727	29.581	29.999	28.264	28.615
家計 HA	159.086	178.515	178.345	164.951	164.986	170.708	170.389	164.543	164.361
商品 CA	47.108	49.229	49.849	46.747	47.350	49.126	49.850	46.739	47.345
商品 CA	88.703	96.670	96.568	92.885	92.834	96.780	96.656	92.886	92.834
商品 CB									
中間需要 QINT	80.757	85.881	86.482	81.352	81.954	86.109	86.810	81.358	81.971
企業 FA	48.208	51.267	51.626	48.563	48.923	51.403	51.821	48.567	48.933
商品 CB	52.876	54.691	54.561	52.181	52.098	54.627	54.475	52.175	52.087
企業 FB	95.452	98.728	98.492	94.197	94.047	98.612	98.337	94.185	94.027
商品 CB									
投資需要 QINV	26.919	26.919	26.919	26.919	26.919	29.232	29.264	27.034	27.096
商品 CA	81.953	81.953	81.953	81.953	81.953	88.965	89.092	82.303	82.491
商品 CB									

GAMS による応用一般均衡分析：海外部門のあるモデル

輸入量 QM	商品 CB	144.000	165.062	166.029	159.343	160.296	164.874	166.004	159.326	160.284
国内供給量 QQ	商品 CA	249.000	260.252	261.737	248.038	249.546	261.163	262.895	248.067	249.613
	商品 CB	538.000	571.731	571.583	547.149	547.342	571.066	570.893	547.082	547.245
国内生産量 QX	商品 CA	279.000	296.705	298.780	281.056	283.138	297.491	299.911	281.075	283.197
	商品 CB	394.000	407.522	406.550	388.820	388.200	407.046	405.911	388.771	388.118
平均要素価格 WF	資本 K	1.000	1.000	1.102	1.000	1.053	1.000	1.103	1.000	1.053
	労働 L	1.180	1.180	1.180	1.255	1.254	1.180	1.180	1.255	1.254
要素の歪み係数 WFDIST	企業 FA	1.000	1.132	1.000	1.080	1.000	1.138	1.000	1.080	1.000
	資本 K	0.610	0.610	0.610	0.610	0.610	0.610	0.610	0.610	0.610
	労働 L	1.000	1.080	1.000	1.032	1.000	1.077	1.000	1.320	1.000
	企業 FB	1.780	1.780	1.780	1.780	1.780	1.780	1.780	1.780	1.780
家計要素所得 YF	家計 HA	125.000	137.292	137.749	131.098	131.599	137.323	137.853	131.092	131.596
	家計 HB	95.000	104.616	104.874	99.887	100.184	104.688	104.984	99.884	100.184
	企業 KA	83.000	91.162	91.465	87.049	87.382	91.183	91.535	87.045	87.380
	企業 KB	82.000	90.300	90.523	86.218	86.475	90.362	90.618	86.216	86.475
政府収入 YG	家計 HA	109.000	111.500	111.507	106.933	107.005	111.442	114.450	106.923	106.993
	家計 HB	285.000	302.450	302.971	290.987	291.618	302.523	303.125	290.976	291.610
	企業 KA	186.000	200.679	201.127	192.268	192.790	200.749	201.268	192.261	192.786
名目 GDP	総額 (支出面、市場価格)	454.000	494.924	496.194	472.471	473.934	495.054	496.525	472.448	473.919
	民間消費	336.000	368.754	369.601	345.344	346.433	359.170	359.976	344.835	345.667
	政府消費	80.000	79.752	79.833	79.681	79.761	79.732	79.827	79.680	79.760
	投資	113.000	113.058	113.039	113.075	113.056	122.737	122.886	113.558	113.798
	輸出	30.000	40.302	40.707	36.041	36.475	40.147	40.620	36.028	36.462
	輸入	-105.000	-106.941	-106.985	-101.669	-101.792	-106.732	-106.785	-101.652	-101.768
	関税率	69.000	71.553	71.584	68.218	68.294	71.499	71.534	68.211	68.285
総額 (要素価格)	総額 (所得面、市場価格)	385.000	423.371	424.611	404.253	405.639	423.556	424.990	404.237	405.634
	総額 (所得面、市場価格)	454.000	494.924	496.194	472.471	473.934	495.054	496.525	472.448	473.919

GAMS による応用一般均衡分析：海外部門のあるモデル

	ベンチマーク 均 衡	仮 設										均 衡						
		i	j	k	l	m	n	o	p									
パラメーター																		
輸出品国際価格 (外貨建て)																		
商品 CA	1.000	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250
要素費用 wfa																		
資本 k	1.000	1.168	1.049	1.120	1.004	1.153	1.049	1.049	1.049	1.049	1.049	1.049	1.049	1.049	1.049	1.049	1.049	1.008
企業 FA	1.000	0.978	1.049	0.930	1.004	0.987	1.004	0.987	1.004	0.987	1.004	0.945	1.008	0.945	1.008	0.945	1.008	1.008
企業 FB	1.000	0.720	0.720	0.761	0.758	0.720	0.758	0.720	0.758	0.720	0.758	0.720	0.758	0.720	0.758	0.720	0.758	0.755
労働 L	2.100	2.100	2.100	2.218	2.212	2.100	2.100	2.212	2.100	2.100	2.100	2.208	2.202	2.100	2.202	2.100	2.202	2.202
企業 FB	2.100	2.100	2.100	2.218	2.212	2.100	2.100	2.212	2.100	2.100	2.100	2.208	2.202	2.100	2.202	2.100	2.202	2.202
変数																		
政府支出 EG	110.000	109.883	110.265	109.838	110.242	109.927	110.265	109.906	110.244	109.927	110.265	109.906	110.244	109.927	110.265	109.906	110.244	110.244
為替レート EXR	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000
海外部門貯蓄 FSAV (外貨建て)	4.000	-25.308	-33.606	-26.940	-35.024	-24.948	-32.032	-26.291	-32.917	-24.948	-32.032	-26.291	-32.917	-24.948	-32.032	-26.291	-32.917	-32.917
投資調整係数 IADJ	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	0.651
限界貯蓄性向 MPS																		
家計 HA	0.264	0.353	0.384	0.394	0.424	0.264	0.264	0.264	0.424	0.264	0.264	0.264	0.264	0.264	0.264	0.264	0.264	0.264
家計 HB	0.221	0.211	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221	0.221
企業価格 PA																		
企業 FA	1.000	1.046	1.004	1.051	1.007	1.041	1.004	1.043	1.007	1.041	1.004	1.043	1.007	1.041	1.004	1.043	1.007	1.007
企業 FB	1.000	0.994	1.015	0.991	1.013	0.996	1.015	0.995	1.013	0.996	1.015	0.995	1.013	0.996	1.015	0.995	1.013	1.013
国産品の国内価格 PD																		
商品 CA	1.000	1.015	0.965	1.021	0.968	1.010	0.965	1.012	0.968	1.010	0.965	1.012	0.968	1.010	0.965	1.012	0.968	0.968
商品 CB	1.000	0.994	1.015	0.991	1.013	0.996	1.015	0.995	1.013	0.996	1.015	0.995	1.013	0.996	1.015	0.995	1.013	1.013
輸出価格 PE (自国通貨建て)																		
商品 CA	1.000	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250	1.250
商品 CB	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000
輸入価格 PM (自国通貨建て)																		
商品 CA	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000
商品 CB	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000	1.000
国内供給価格 PQ																		
商品 CA	1.040	1.056	1.004	1.062	1.007	1.050	1.004	1.053	1.007	1.050	1.004	1.053	1.007	1.050	1.004	1.053	1.007	1.007
商品 CB	1.037	1.032	1.048	1.030	1.047	1.034	1.048	1.033	1.047	1.034	1.048	1.033	1.047	1.034	1.048	1.033	1.047	1.047

GAMS による応用一般均衡分析：海外部門のあるモデル

輸入量 QM	144.000	141.991	140.795	135.327	134.563	142.743	141.866	136.956	136.449
商品 CB									
国内供給量 QQ	249.000	253.412	258.468	242.038	247.646	251.365	255.328	240.191	244.599
商品 CA									
商品 CB	538.000	532.259	522.127	507.921	499.340	534.409	526.098	513.047	506.311
国内生産量 OX	279.000	301.306	313.440	287.180	299.925	299.500	309.632	285.901	296.268
商品 CA									
商品 CB	394.000	390.269	381.339	372.597	364.783	391.667	984.239	376.092	369.868
平均要素価格 WF	1.000	1.000	1.049	1.000	1.004	1.000	1.090	1.000	1.008
資本 K									
労働 L	1.180	1.180	1.180	1.247	1.243	1.180	1.180	1.241	1.237
要素の歪み係数 WFDIST									
企業 FA	1.000	1.168	1.000	1.120	1.000	1.153	1.000	1.104	1.000
資本 K	0.610	0.610	0.610	0.610	0.610	0.610	0.610	0.610	0.610
労働 L	1.000	0.978	1.000	0.930	1.000	0.987	1.000	0.945	1.000
企業 FB	1.000	1.780	1.000	1.780	1.780	1.780	1.780	1.780	1.780
労働 L									
家計要素所得 YF	125.000	130.605	131.182	124.556	125.480	130.640	131.182	125.132	125.965
家計 HA									
労働 L	95.000	100.262	100.529	95.670	96.192	100.172	100.415	95.944	96.416
資本 K	83.000	86.721	87.105	82.705	83.319	86.745	87.105	83.088	83.641
労働 L	82.000	86.542	86.772	82.578	83.029	86.464	86.674	82.815	83.222
政府収入 YG	109.000	109.467	108.830	105.359	104.973	109.638	109.137	105.957	105.689
家計所得 YH	285.000	295.867	296.711	285.226	286.672	295.811	296.597	286.076	287.381
家計 HB	186.000	194.263	194.877	186.283	187.348	194.209	194.779	186.902	187.863
名目 GDP	454.000	472.612	473.359	45.845	452.839	472.679	473.463	452.835	454.716
総額 (支出面、市場価格)									
民間消費	336.000	325.394	317.550	302.029	295.631	349.620	350.590	337.420	339.041
政府消費	80.000	79.883	80.265	79.838	80.242	79.927	80.265	79.906	80.244
投資	113.000	113.028	112.938	113.038	112.943	89.184	81.576	80.219	73.513
輸出	30.000	57.844	65.269	54.616	62.142	58.031	64.476	55.154	61.412
輸入	-105.000	-103.535	-102.663	-98.676	-98.119	-104.083	-103.444	-99.864	-99.494
間接税	69.000	68.482	67.770	65.336	64.819	68.859	68.087	65.857	65.472
総額 (要素価格)	385.000	404.130	405.589	385.509	388.020	404.020	405.376	386.978	389.244
総額 (所得面、市場価格)	454.000	472.612	473.359	450.845	452.839	472.679	473.463	452.835	454.716

表 8 仮設均衡の社会勘定行列

類型 (a)	企業		商品		資本		労働		家計			政府	貯蓄-投資	所得税	売上税	関税	海外部門	合計	
	FA	FB	CA	CB	K	L	HA	HB	GOV	S-I	Y-TAX								STAX
企業	92.229	58.733	308.997	419.351														308.997	419.351
商品	52.641	101.374					33.327	52.867	13.422	28.908							40.302	319.788	
資本	82.629	145.825					183.299	99.261	66.330	84.150								587.055	
労働	81.498	113.419																228.454	
家計					137.292	104.616												194.317	
政府					91.162	90.300			25.000	5.000								302.449	
貯蓄-投資												1.749						14.216	200.678
所得税														26.619	31.832	39.721		13.328	111.500
消費税																			
関税																			
海外部門																			
合計	308.997	419.351	319.788	587.054	228.454	194.916	302.450	200.680	111.501	113.058	26.619	31.832	39.721	106.941				106.941	

行和と列和の差 0.000 0.000 0.000 0.001 0.000 0.001 -0.001 -0.002 -0.001 0.001 0.001 0.001 0.000 0.000 0.000 0.000 0.000

示されている。

7. 残された課題

本稿は、小平 (2002a) 以来進めてきた数理計画法解法プログラム GAMS を利用した応用一般均衡モデル構築の練習の最終回にあたる。現実の経済の応用政策分析の出発点となるようなモデルを構築すべく、われわれは基本モデル age 1 を段階的に拡張して、生産における中間投入、要素市場のより一般的な取扱い、政府、海外部門を取り入れてきた。

しかし、経済構造を一層よく反映したモデルにするためには、まだいくつかの課題が残されている。財市場を需給をバランスさせるように、価格が伸縮的に調整されるという前提を維持しながら、モデルの検討を進めてきたが、現実の経済では価格硬直性や価格統制が存在する場合もあり、これらの現象の定式化は手つかずのまま残されている。また、生産関数や効用関数の関数形については、計算の便宜から Cobb-Douglas 型を仮定してきた。これは、生産および消費において代替の弾力性の値を 1 と想定することと同値である。代替の弾力性を 1 とおくことはかなり制約的な仮定であることが知られているので、1 以外の値を想定できるように、生産関数や効用関数について自由な（したがって取扱いが複雑になる）関数形を利用することも検討すべきである。最後に、貿易の定式化においても、小国の仮定を外すことを考えることも必要である。

参 考 文 献

小平裕 (2002a) 「GAMS による応用一般均衡分析：基本モデル」『経済研究』156号, 2002年3月。

小平裕 (2002b) 「GAMS による応用一般均衡分析：中間投入のあるモデル」『経済研究』157号, 2002年6月。

7) 紙幅を節約するために表 8 には類型 (a) の社会勘定行列のみを掲げる。

GAMS による応用一般均衡分析：海外部門のあるモデル

小平裕 (2002c) 「GAMS による応用一般均衡分析：資本市場と企業固有の生産要素」『経済研究』158号, 2002年11月。

小平裕 (2003) 「GAMS による応用一般均衡分析：政府活動と失業」『経済研究』159号, 2003年1月。

Armington, P. S., (1969), “A Theory of Demand for Products Distinguished by Place of Production,” *International Monetary Fund Staff Paper*, vol. 16 no. 1.

Brooke, A., Kendrick, D., and Meeraus, A., (1988), *GAMS: A User's Guide Release 2. 25*, Boyd and Fraser.

Lofgren, H., (1999), *Exercises in General Equilibrium Modeling using GAMS*, International Food Policy Research Institute.